HEWLETT-PACKARD

HP-25

Programmes d'applications



HEWLETT hp PACKARD

172 points de vente dans 65 pays assurent le service après-vente.

Hewlett-Packard France:
Siège social: Quartier de Courtabœuf, boîte postale n° 6, 91401 Orsay, tél. (1) 907 78 25
Agence de Lille: Centre Vauban, 201, rue Colbert, Entrée A2, 59000 Lille, tél. (20) 51 44 14
Agence de Lyon: Chemin des Mouilles, boîte postale n° 12, 69130 Ecully, tél. (78) 33 81 25
Agence de Marseille: Aéroport principal de Marseille-Marignane, 13721 Marignane, tél. (91) 89 12 36
Agence de Rennes: 63, avenue de Rochester, 35000 Rennes, tél. (99) 36 33 21
Agence de Strasbourg: 74, allée de la Robertsau, 67000 Strasbourg, tél. (88) 35 23 20/21
Agence de Toulouse: Péricentre de la Cépière, chemin de la Cépière, 31300 Toulouse Le Mirail, tél. (61) 40 11 12

Pour la Belgique: Hewlett-Packard Benelux S.A., 1, avenue du Col-Vert, B-1170 Bruxelles, tél. (02/03) 672 22 40

Pour la Suisse romande: Hewlett-Packard (Schweiz) AG, 9, chemin Louis-Pictet, 1214 Vernier-Genève, tél. (022) 41 49 57

Pour les pays du bassin méditerranéen, Afrique du Nord et Moyen-Orient: 35, Kolokotroni Street – Platia Kefallariou, GR-Kifissia-Athènes, Grèce, tél. 80 80 337/359/429, 80 81 741/742/743/744 et 80 18 693

Pour l'Autriche/Pour les pays socialistes: Hewlett-Packard Ges.m.b.H., Handelskai 52/53, boîte postale n° 7, A-1205 Vienne, Autriche, tél. (0222) 35 16 21 à 32

Pour l'URSS: Hewlett-Packard Representative Office USSR, Hotel Budapest, Room 201, Petrovskie Linii 2/18, 103–051 Moscow

Pour le Canada: Hewlett-Packard (Canada) Ltd., 275 Hymus Boulevard, Pointe-Claire H9R 1G7, tél. (514) 697-4232
Hewlett-Packard (Canada) Ltd., 2376 Galvani, Ste-Foy G1N 4G4, tél. (418) 688-8710

Direction pour l'Europe: Hewlett-Packard S.A., 7, rue du Bois-du-Lan, boîte postale n° 349, CH-1217 Meyrin 1-Genève, Suisse, tél. (022) 41 54 00

Printed in Singapore

00025–90013 French

INTRODUCTION

Les programmes figurant dans ce fascicule ont été choisis dans le domaine des mathématiques, des statistiques, de la finance, de la topographie, de la navigation et des jeux; ils sont regroupés en 8 chapitres.

Chaque programme est présenté de la manière suivante: description générale, formules utilisées, listing, identification des registres mémoire utilisés, mode opératoire et résolution de 1 ou 2 exemples numériques. Pour utiliser les programmes, il n'est pas nécessaire d'être un expert en programmation: il suffit simplement de lire attentivement le manuel d'utilisation du HP-25: les programmes présentés vous permettront ensuite d'accroître vos connaissances sur les principes et les techniques de la programmation.

Le premier programme de chaque chapitre, en plus de la présentation indiquée ci-dessus, contient une description plus détaillée du problème, un listing commenté des touches utilisées lors de la programmation avec le contenu pas à pas des registres de la pile opérationnelle, et les pressions de touches nécessaires à la résolution du problème.

Chaque fois qu'une technique de programmation intéressante aura été utilisée, elle vous sera indiquée dans un paragraphe intitulé «Remarques sur la programmation», précédant immédiatement le listing des touches utilisées dans la rédaction du programme.

Que votre intérêt réside dans la résolution de problèmes particuliers d'un domaine spécifique ou dans la volonté d'en savoir plus sur la puissance de programmation de votre calculateur, nous espérons que ce fascicule vous aidera à utiliser au maximum votre HP-25.

SOMMAIRE

UN MOT AU SUJET DE LA PROGRAMMATION

Ce fascicule contient les informations nécessaires pour l'utilisation de chaque programme. En plus d'un bref exposé du problème, d'une liste des formules utilisées et de la résolution d'un exemple numérique, il existe deux tableaux: feuille de programmation et mode opératoire.

Feuille de programmation

La feuille de programmation détaillée est utilisée seulement dans le premier programme de chaque chapitre:

AFFICHAGE		TOUCHES	X	Y	Z	T	COMMENTAIRES	REGISTRES
PAS	CODE							
00			v	θ			Conversion polaire/rectangu-	R_0 Δt
01	14 09	f →R	v_x	v_y			laire de $v_x = v \cos\theta$	
02	23 02	STO 2	v_x	v_y				
03	21	x⇄y	v_y	v_x			$v_y = v \sin\theta$	R_1 g
04	23 03	STO 3	v_y	v_x				
05	00	0	0					
06	23 04	STO 4	0				Initialiser t=0	
07	24 00	RCL 0	Δt				Incrémentation	R_2 v_x
08	23 51 04	STO + 4	Δt				Intervalle du temps suivant	
09	24 04	RCL 4	t				$t \leftarrow t + \Delta t$	R_3 v_y
10	15 02	g x^2	t^2					

Feuille de programmation détaillée (calcul d'une courbe point par point – Chapitre 1).

Les deux premières colonnes indiquent les codes affichés lors de l'introduction du programme:
- la colonne PAS donne le numéro de pas occupé par l'instruction;
- la colonne CODE donne le code numérique de la touche pressée;
- la colonne TOUCHES donne la séquence de touches nécessaires à la
- rédaction du programme. La touche **ENTER↑** est représentée dans cette colonne par ↑. Toutes les autres touches sont désignées par le symbole qui est le leur sur le clavier;
- les quatre colonnes X, Y, Z, T indiquent les contenus des quatre registres de la pile opérationnelle après chaque pression de touche;
- la colonne COMMENTAIRES fournit des explications supplémentaires pour les calculs du programme;
- la colonne de droite REGISTRES indique les données qui ont été stockées dans les registres mémoire R_0 à R_7.

Les colonnes X, Y, Z, T et COMMENTAIRES vous permettent de mieux suivre le déroulement d'un programme et d'augmenter vos connaissances techniques de programmation.
La feuille de programmation simplifiée est semblable à la feuille de programmation détaillée, mais elle ne comporte pas les colonnes X, Y, Z, T et COMMENTAIRES.

Mode opératoire

Le mode opératoire sert de guide pour l'utilisation des programmes, et se présente sous la forme d'un tableau comprenant 5 colonnes. L'exemple ci-après décrit le mode opératoire du programme «Calcul d'une courbe point par point» (Chapitre 1).

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire l'incrément de temps	Δt	STO	0			
3	Mettre en mémoire la constante de gravité	g	STO	1			
4	Introduire l'angle et la vitesse initiale	θ	↑				
		v	f	PRGM			
5	Effectuer 5 et 6 pour chaque point						
	Affichage du temps et de la distance horizontale		R/S				(t)
							x
6	Affichage de la hauteur		R/S				y
7	Pour un autre θ ou v,						
	aller en 4. Pour un						
	autre Δ t ou g, aller						
	en 2 ou 3, puis en 4.						

- la colonne NUMÉRO indique l'ordre séquentiel des opérations à effectuer,
- la colonne INSTRUCTIONS indique les instructions et les commentaires relatifs aux opérations à exécuter. Les instructions sont exécutées séquentiellement, sauf indication contraire donnée dans cette colonne.

Normalement, la première instruction est «Introduire le programme», c'est-à-dire mettre en mémoire la séquence de touches du programme (passer en mode PRGM, appuyer sur les touches f PRGM, introduire le programme, et revenir en mode RUN).

Lorsqu'une série d'instructions est à répéter, elle est entourée d'un cadre imprimé en gras (dans cet exemple, les instructions 5 et 6 sont répétées afin d'obtenir un nombre de paires (x, y) pour un graphe).

- la colonne DONNÉES indique les données à introduire et leurs unités.

- la colonne TOUCHES indique les touches à presser. ↑ est le symbole de la touche **ENTER↑**. Toutes les autres touches sont désignées par le symbole qui est le leur sur le clavier. Ne pas tenir compte des cases laissées en blanc dans cette colonne.

Certains programmes plus complexes nécessitent la pression de plusieurs touches avant que le calculateur n'affiche de résultat. Dans ce cas, elles sont indiquées dans la colonne TOUCHES.

la colonne RÉSULTATS donne tous les résultats, intermédiaires ou définitifs, calculés soit à partir du clavier, soit par l'exécution du programme.
Si une variable est placée entre parenthèses (par exemple (t) à l'instruction 5), cela signifie que le résultat peut être affiché momentanément par une instruction PAUSE ([f] [PAUSE]).

CHAPITRE 1 : ALGÈBRE ET THÉORIE DES NOMBRES

CALCUL D'UNE COURBE POINT PAR POINT

Rien n'est plus ennuyeux que d'étudier les variations d'une fonction. Parfois même c'est un exercice bien difficile si le degré de l'équation est élevé. Le tracé de la parabole $y=3x^2-4x+4$, pour des valeurs entières de x comprises entre $-\infty$ et $+\infty$, n'est guère plus amusant. Un calculateur programmable tel que le HP-25 est un outil bien pratique pour préparer le tracé d'un graphe.
Il permet d'obtenir des couples (x, y) en mémorisant le programme calculant y pour x donné. Il suffit ensuite de revenir en début de mémoire, d'introduire une valeur de x, puis de presser la touche [R/S]. Ces opérations seront répétées pour chaque valeur de x.

Un pas supplémentaire inséré dans le programme permet de calculer automatiquement les y correspondants à des x tabulés, c'est-à-dire tels que x_1, $x_1+\Delta x$, $x_1+2\Delta x$, ... avec Δx donné.
Ci-dessous est représenté l'organigramme :



Le programme décrit dans ce fascicule pour illustrer cette méthode est une extension de ce type général de problème. Il a pour but de représenter graphiquement la trajectoire d'une pierre projetée avec une vitesse initiale v et à un angle θ par rapport à l'horizontale. La résistance de l'air étant négligée, les équations suivantes donnent les coordonnées x et y de la pierre en fonction du temps t:

$$x = vt \cos\theta$$

$$y = vt \sin\theta - \frac{1}{2} gt^2$$

où x: distance horizontale atteinte par la pierre
y: hauteur atteinte par la pierre
g: constante de gravité ($g \simeq 9{,}8\ m/s^2$)

Ces équations paramétrées sont légèrement différentes des équations classiques dans lesquelles y est une fonction de x; ici, x et y sont tous deux fonctions d'un paramètre t. Les points à représenter sur le graphe sont toujours les couples (x, y). Dans cet exemple, le temps r est incrémenté selon une progression arithmétique (Δ t constant).

Remarques:

1. N'importe quel système d'unité peut être utilisé.
2. Il n'y a pas de programme général effectuant le calcul d'une courbe point par point; la méthode décrite précédemment permet de résoudre un type de problème. Toutefois, le listing des touches et l'organigramme vous permettront de modifier facilement ce programme afin de l'adapter à votre propre problème.

Remarques sur la programmation:

1. Les composantes v_x et v_y du vecteur vitesse sont calculées au moyen d'un seul pas de programme, v et θ étant convertis en coordonnées rectangulaires ($\boxed{f}\boxed{\to R}$). Les valeurs $v_x = v \cos\theta$ et $v_y = v \sin\theta$ se trouvent respectivement dans les registres X et Y.
2. Ce programme contient une instruction PAUSE ($\boxed{f}$ $\boxed{\text{PAUSE}}$) qui permet d'afficher pendant 1 seconde environ la variable t (0.25 – 0.50 – 0.75, etc.).

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00			v	θ			Conversion polaire/rectangu-
01	14 09	f →R	v_x	v_y			laire de $v_x = v \cos\theta$
02	23 02	STO 2	v_x	v_y			
03	21	x⇄y	v_y	v_x			$v_y = v \sin\theta$
04	23 03	STO 3	v_y	v_x			
05	00	0	0				
06	23 04	STO 4	0				Initialiser t=0
07	24 00	RCL 0	Δt				Incrémentation
08	23 51 04	STO + 4	Δt				Intervalle du temps suivant
09	24 04	RCL 4	t				$t \leftarrow t + \Delta t$
10	15 02	g x^2	t^2				
11	24 01	RCL 1	g	t^2			
12	61	x	$g\,t^2$				
13	02	2	2	$g\,t^2$			
14	71	÷	$1/2\,g\,t^2$				
15	32	CHS	$-1/2\,g\,t^2$				
16	24 04	RCL 4	t	$-1/2\,g\,t^2$			
17	24 03	RCL 3	v_y	t	$-1/2\,g\,t^2$		
18	61	x	$v_y\,t$	$-1/2\,g\,t^2$			
19	51	+	y				$y = v_y t - ½\,gt^2$
20	24 04	RCL 4	t	y			
21	24 02	RCL 2	v_x	t	y		
22	61	x	x	y			$x = v_x\,t$
23	24 04	RCL 4	t	x	y		
24	14 74	f PAUSE	t	x	y		Affichage momentané de t
25	22	R↓	x	y		t	
26	74	R/S	x	y		t	Affichage de x
27	21	x⇄y	y	x		t	
28	74	R/S	y	x		t	Affichage de y
29	13 07	GTO 07	y	x		t	Branchement au pas 7 pour
30							autre t.
31							
32							
33							
34							
35							
36							
37							
38							
39							
40							
41							
42							
43							
44							
45							
46							
47							
48							
49							

REGISTRES	
R_0	Δt
R_1	g
R_2	v_x
R_3	v_y
R_4	t
R_5	
R_6	
R_7	

Exemple:

Tracer la trajectoire d'une pierre projetée avec une vitesse de 20 m/s et à un angle de 30° par rapport à l'horizontale.
Intervalle de temps entre les points à calculer: 0.25 seconde. Constante de gravité $g = 9.8$ m/s².

Solution:

0.25 STO 0 9.8 STO 1 30 ↑ 20 f PRGM R/S ⟶ 0.25 (t_1)
4.33 (x_i)
R/S ⟶ 2.19 (y_1)
R/S ⟶ 0.5 (t_2)
8.66 (x_2)
R/S ⟶ 3.78 (y_2)
R/S ⟶ 0.75 (t_3)
12.99 (x_3)
R/S ⟶ 4.74 (y_3)

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire l'incrément de temps	Δt	STO	0			
3	Mettre en mémoire la constante de gravité	g	STO	1			
4	Introduire l'angle et la	θ	↑				
	vitesse initiale	v	f	PRGM			
5	Effectuer 5 et 6 pour chaque point						
	Affichage du temps et de la distance horizontale		R/S				(t)
							x
6	Affichage de la hauteur		R/S				y
7	Pour un autre θ ou v,						
	aller en 4. Pour un						
	autre Δ t ou g, aller						
	en 2 ou 3, puis en 4.						

Continuer à presser la touche R/S jusqu'au moment où une valeur négative de y est obtenue. Ci-dessous est donné le tableau des résultats:

t	0.25	0.50	0.75	1.00	1.25	1.50	1.75	2.00	2.25
x	4.33	8.66	12.99	17.32	21.65	25.98	30.31	34.64	38.97
y	2.19	3.78	4.74	5.10	4.84	3.98	2.49	0.40	-2.31

La trajectoire de la pierre est une parabole.



ÉQUATION DU SECOND DEGRÉ

Les racines x_1, x_2 de l'équation

$$ax^2 + bx + c = 0$$

sont:

$$x_{1,2} = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}$$

Pour obtenir une bonne précision, calculez d'abord la racine de plus grande valeur absolue au moyen de la formule suivante:

$$x_1 = \frac{-ab}{|ab|}\left(\left|\frac{b}{2a}\right| + \sqrt{\frac{b^2-4ac}{4a^2}}\right)$$

puis l'autre racine $$x_2 = \frac{c}{x_1 a}$$

Si le discriminant $D = (b^2 - 4ac)/4a^2$
est positif ou nul, les racines sont réelles. Sinon, elles sont imaginaires conjuguées et égales à:

$$u \pm iv = \frac{-b}{2a} \pm \frac{\sqrt{4ac - b^2}}{2a}\, i$$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	31	↑
02	22	R↓
03	71	÷
04	02	2
05	71	÷
06	32	CHS
07	31	↑
08	15 02	g x^2
09	22	R↓
10	22	R↓
11	21	x⇄y
12	71	÷
13	23 00	STO 0
14	41	–
15	14 74	f PAUSE
16	15 41	g x<0
17	13 31	GTO 31
18	14 02	f $\sqrt{x}$
19	21	x⇄y
20	15 41	g x<0
21	13 24	GTO 24
22	51	+
23	13 26	GTO 26
24	21	x⇄y

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	41	–
26	74	R/S
27	15 22	g 1/x
28	24 00	RCL 0
29	61	x
30	13 00	GTO 00
31	32	CHS
32	14 02	f $\sqrt{x}$
33	21	x⇄y
34	74	R/S
35	21	x⇄y
36	13 00	GTO 00
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 c/a
R_1
R_2
R_3
R_4
R_5
R_6
R_7

Exemples:

1. $x^2+x-6=0$
2. $3x^2+2x-1=0$
3. $2x^2-3x+5=0$

Solutions:

1. $D=6.25$
   $x_1=-3.00$
   $x_2=2.00$
2. $D=0.44$
   $x_1=-1.00$
   $x_2=0.33$
3. $D=-1.94$
   $x_1, x_2=0.75 \pm 1.39\ i$

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Retourner en début de programme		f	PRGM			
3	Introduire les coefficients et démarrer le calcul; D s'affiche	c	↑				
	momentanément	b	↑				
		a	R/S				(D)
4	Si D≥0, racines réelles						x_1
	ou		R/S				x_2
	si D < 0, racines complexes						
	de la forme u ±iv						
							u
			R/S				v
5	Pour un nouveau cas, aller en 2.						

OPÉRATIONS (+, −, ×, ÷) SUR DES NOMBRES COMPLEXES

Soient a_1+ib_1 et a_2+ib_2 deux nombres complexes. Les opérations arithmétiques $+, -, \times, \div$ sont définies comme suit:

1. addition (+) $(a_1+ib_1)+(a_2+ib_2)=(a_1+a_2)+(b_1+b_2)i$
2. soustraction (−) $(a_1+ib_1)-(a_2+ib_2)=(a_1-a_2)+(b_1-b_2)i$
3. multiplication (×) $(a_1+ib_1)\times(a_2+ib_2)=r_1 r_2\, e^{i(\theta_1+\theta_2)}$
4. division (÷) $\dfrac{(a_1+ib_1)}{(a_2+ib_2)}=\dfrac{r_1}{r_2}\, e^{i(\theta_1-\theta_2)},\ a_2+ib_2\neq 0$

où $r_1 e^{i\theta_1}$ est la représentation polaire de a_1+ib_1 et $r_2 e^{i\theta_2}$ la représentation polaire de a_2+ib_2. Dans chaque cas, la réponse sera de la forme $x+iy$.

Après l'exécution d'un calcul, x est stocké dans les registres R_1 et X, y dans les registres R_2 et Y: des opérations arithmétiques peuvent ainsi être effectuées en chaîne.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	32	CHS
02	21	x⇄y
03	32	CHS
04	21	x⇄y
05	24 00	RCL 0
06	51	+
07	21	x⇄y
08	24 01	RCL 1
09	51	+
10	13 31	GTO 31
11	15 09	g →P
12	15 22	g 1/x
13	21	x⇄y
14	32	CHS
15	21	x⇄y
16	13 18	GTO 18
17	15 09	g →P
18	24 02	STO 2
19	22	R↓
20	24 01	RCL 1
21	24 00	RCL 0
22	15 09	g →P
23	24 02	RCL 2
24	61	x

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	23 02	STO 2
26	22	R↓
27	51	+
28	24 02	RCL 2
29	14 09	f →R
30	21	x⇄y
31	23 01	STO 1
32	21	x⇄y
33	23 00	STO 0
34	13 00	GTO 00
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 a_1, x
R_1 b_1, y
R_2 Utilisé
R_3
R_4
R_5
R_6
R_7

Exemples:

1. $(1.2+3.7i)-(2.6-1.9i)=-1.4+5.6i$

2. $\dfrac{3+4i}{7-2i}=0.25+0.64i$

3. $\left[\dfrac{(3+4i)+(7.4-5.6i)}{(7-2i)}\right][3.1+4.6i]=3.61+7.16i$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire le	b_1	STO	1			
	premier nombre complexe	a_1	STO	0			
3	Introduire le second	b_2	↑				
	nombre	a_2					
4	Pour une addition		GTO	05	R/S		x
	ou						
	pour une soustraction		f	PRGM	R/S		x
	ou						
	pour une multiplication		GTO	17	R/S		x
	ou						
	pour une division		GTO	11	R/S		x
5	Pour la partie imaginaire		x⇄y				y
6	Pour le calcul en chaîne						
	suivant, aller en 3						
7	Pour un nouveau calcul, aller en 2						

FONCTIONS D'UNE VARIABLE COMPLEXE
$|z|$, z^2, $1/z$, $\sqrt{z}$

Un nombre complexe $z=a+ib$ a la représentation polaire $re^{i\theta}$. Les formules servant aux calculs de ces fonctions sont les suivantes:

1. $|z|=r$
2. $z^2=r^2\,e^{i2\theta}$
3. $1/z=\dfrac{1}{r}\,e^{-i\theta},\ z\neq 0$
4. $\sqrt{z}=\pm(\sqrt{r}\,e^{i\theta/2})=\pm(x+iy)$

La réponse est de la forme $x+iy$.

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	15 09	g →P
02	13 00	GTO 00
03	15 09	g →P
04	15 02	g x^2
05	21	x⇄y
06	31	↑
07	51	+
08	21	x⇄y
09	14 09	f →R
10	13 00	GTO 00
11	15 09	g →P
12	15 22	g 1/x
13	21	x⇄y
14	32	CHS
15	21	x⇄y
16	14 09	f →R
17	13 00	GTO 00
18	15 09	g →P
19	14 02	f $\sqrt{x}$
20	21	x⇄y
21	02	2
22	71	÷
23	21	x⇄y
24	14 09	f →R

AFFICHAGE		TOUCHES
PAS	CODE	
25	13 00	GTO 00
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1
R_2
R_3
R_4
R_5
R_6
R_7

Exemples:

1. $|12-5i| = 13.00$
2. $(6-i)^2 = 35.00-12.00i$
3. $\dfrac{1}{2+5i} = 0.07-0.17i$
4. $\sqrt{3+4i} = \pm(2.00+1.00i)$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire z	b	↑				
		a					
3	Pour $\|z\|$		f	PRGM	R/S		$\|z\|$
	ou						
	z^2		GTO	03	R/S		x
			x⇄y				y
	ou						
	1/z		GTO	11	R/S		x
			x⇄y				y
	ou						
	$\sqrt{z}$		GTO	18	R/S		x
			x⇄y				y
4	Pour un nouveau cas, aller en 2.						

DÉTERMINANT ET INVERSE D'UNE MATRICE 2×2

Soit $A = \begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix}$ une matrice 2 × 2.

Le déterminant de la matrice A (Det A ou $|A|$) est égal à:
Det A $= a_{22}a_{11} - a_{12}a_{21}$
En outre, le programme calcule l'inverse A^{-1} de A au moyen de la formule suivante:

$$A^{-1} = \begin{bmatrix} a_{22}/\text{Det A} & -a_{12}/\text{Det A} \\ -a_{21}/\text{Det A} & a_{11}/\text{Det A} \end{bmatrix}$$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	24 04	RCL 4
02	24 01	RCL 1
03	61	x
04	24 02	RCL 2
05	24 03	RCL 3
06	61	x
07	41	−
08	23 00	STO 0
09	74	R/S
10	24 04	RCL 4
11	24 00	RCL 0
12	71	÷
13	74	R/S
14	24 02	RCL 2
15	24 00	RCL 0
16	71	÷
17	32	CHS
18	74	R/S
19	24 03	RCL 3
20	24 00	RCL 0
21	71	÷
22	32	CHS
23	74	R/S
24	24 01	RCL 1

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	24 00	RCL 0
26	71	÷
27	13 00	GTO 00
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 Det A
R_1 a_{11}
R_2 a_{12}
R_3 a_{21}
R_4 a_{22}
R_5
R_6
R_7

Exemple:

Calcul du déterminant et de l'inverse de la matrice

$$A = \begin{bmatrix} 3 & 2 \\ 4 & -4 \end{bmatrix}$$

Solution:

Det A = −20

$$A^{-1} = \begin{bmatrix} 0.20 & 0.10 \\ 0.20 & -0.15 \end{bmatrix}$$

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire la matrice	a_{11}	STO	1			
		a_{12}	STO	2			
		a_{21}	STO	3			
		a_{22}	STO	4			
3	Calcul du déterminant		f	PRGM	R/S		Det A
4	Calcul de l'inverse		R/S				a_{11}^{-1}
			R/S				a_{12}^{-1}
			R/S				a_{21}^{-1}
			R/S				a_{22}^{-1}
5	Pour un nouveau cas, aller en 2.						

CONVERSION D'UN NOMBRE EN BASE b EN UN NOMBRE EN BASE 10

Ce programme est constitué de deux sous-programmes. Le premier convertit la partie entière d'un nombre en base b en un nombre en base 10.

$$I_{10} = i_n i_{n-1} \ldots i_2 i_1 = i_n b^{n-1} + i_{n-1} b^{n-2} + \ldots + i_2 b + i_1$$

L'évaluation se fait sous la forme:

$$b\,(\ldots(b(b(i_n b + i_{n-1}) + i_{n-2}) + \ldots) + i_2) + i_1$$

Le second sous-programme convertit la partie fractionnaire d'un nombre en base b en un nombre en base 10.

$$F_{10} = f_1 f_2 \ldots f_m = f_1 b^{-1} + f_2 b^{-2} + \ldots + f_m b^{-m}$$

Ces deux programmes peuvent donc convertir tout nombre en base b en un nombre en base 10. Les zéros doivent être correctement positionnés.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	23 01	STO 1
02	24 00	RCL 0
03	31	↑
04	31	↑
05	31	↑
06	24 01	RCL 1
07	74	R/S
08	23 01	STO 1
09	34	CLX
10	51	+
11	61	x
12	24 01	RCL 1
13	51	+
14	13 07	GTO 07
15	24 00	RCL 0
16	15 22	g 1/x
17	23 02	STO 2
18	23 03	STO 3
19	61	x
20	74	R/S
21	24 02	RCL 2
22	24 03	RCL 3
23	61	x
24	23 03	STO 3

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	61	x
26	51	+
27	13 20	GTO 20
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 b
R_1 Utilisé
R_2 b^{-1}
R_3 b^{-j}
R_4
R_5
R_6
R_7

Exemples:

1. $1777_8 = 1023_{10}$
2. $143.2044_5 = 48.4384_{10}$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre la base en mémoire	b	STO	0			
3	Pour la partie entière, introduire le chiffre le plus à gauche						
		i_n	f	PRGM	R/S		
4	Effectuer 4 pour j=n–1, ... 2 Introduire le chiffre suivant						
		i_j*	R/S				
5	Introduire le dernier chiffre	i_1*	R/S				I_{10}
6	Pour la partie fractionnaire, introduirele chiffre après la virgule						
		f_1	GTO	15	R/S		
7	Effectuer 7 pour j=2, ... n–1						
	Introduire le chiffre suivant	f_j*	R/S				
8	Introduire le dernier chiffre	f_m*	R/S				F_{10}
9	Pour un nouveau cas,						
	aller en 2						
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

CONVERSION D'UN NOMBRE EN BASE 10 EN UN NOMBRE EN BASE b

Ce programme convertit n'importe quel nombre positif exprimé en base 10, N_{10} en un nombre en base b, N_b ($2 \leqslant b \leqslant 100$). Il utilise un algorithme itératif qui à chaque itération augmente de 1 le nombre de digit de N_b. Après chaque itération, le programme s'arrête pendant 1 seconde environ pour afficher des approximations successives de la réponse définitive. Quand la valeur affichée de N_b atteint la précision désirée, presser la touche [R/S] pour arrêter le programme, puis les touches [RCL] [3] pour afficher N_b.

Remarques:

1. Si la base b est telle que $11 \leqslant b \leqslant 100$, chaque digit s'affiche sur l'écran au moyen de 2 chiffres. Par exemple, 4B6.C sera affiché en base 16 comme 41106.12.

2. Si, durant l'exécution du programme, la précision du calculateur est dépassée, le HP-25 donne un résultat incorrect. La valeur de Nb se trouve dans le registre R3.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00			25	24 02	RCL 2
01	24 00	RCL 0	26	21	$x \rightleftarrows y$
02	01	1	27	14 03	f y^x
03	00	0	28	24 03	RCL 3
04	14 51	f $x \geqslant y$	29	51	+
05	13 09	GTO 09	30	23 03	STO 3
06	01	1	31	14 74	f PAUSE
07	00	0	32	14 74	f PAUSE
08	00	0	33	24 00	RCL 0
09	23 02	STO 2	34	24 04	RCL 4
10	00	0	35	14 03	f y^x
11	23 03	STO 3	36	23 41 01	STO – 1
12	24 01	RCL 1	37	13 12	GTO 12
13	14 07	f LN	38		
14	24 00	RCL 0	39		
15	14 07	f LN	40		
16	71	÷	41		
17	15 41	g $x<0$	42		
18	13 21	GTO 21	43		
19	14 01	f INT	44		
20	13 24	GTO 24	45		
21	14 01	f INT	46		
22	01	1	47		
23	41	–	48		
24	23 04	STO 4	49		

REGISTRES
R_0 b
R_1 N_{10}
R_2 10 ou 100
R_3 N_b
R_4 1 digit
R_5
R_6
R_7

Exemples:

1. $67.32_{10} = 403.050114_{16}$
   $= 43.51\,E_{16}$
2. $\pi = 3.141592654_{10} = 11.00100100_2$

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Choisir le format d'affichage		f	FIX	9		
3	Mettre en mémoire la base et le	b	STO	0			
	nombre en base 10	N_{10}	STO	1	f	PRGM	
4	Afficher les approximations						
	successives de N_b		R/S				(N_b)
5	Lorsque le nombre est						
	affiché avec la précision désirée,						
	appuyer sur R/S (arrêt)		RCL	3			N_b
6	Pour un nouveau cas, aller en 3						

PRODUIT VECTORIEL

Si $A = (a_1, a_2, a_3)$ et $B = (b_1, b_2, b_3)$ sont deux vecteurs tridimensionnels, le produit vectoriel de A et B $(A \times B)$ se calcule de la façon suivante:

$$A \times B = \left(\begin{vmatrix} a_2 & a_3 \\ b_2 & b_3 \end{vmatrix}, -\begin{vmatrix} a_1 & a_3 \\ b_1 & b_3 \end{vmatrix}, \begin{vmatrix} a_1 & a_2 \\ b_1 & b_2 \end{vmatrix} \right) = (a_2 b_3 - a_3 b_2, a_3 b_1 - a_1 b_3, a_1 b_2 - a_2 b_1)$$

La solution est de la forme (c_1, c_2, c_3).

AFFICHAGE		TOUCHES	AFFICHAGE		TOUCHES
PAS	CODE		PAS	CODE	
00			25		
01	24 02	RCL 2	26		
02	24 06	RCL 6	27		
03	61	x	28		
04	24 03	RCL 3	29		
05	24 05	RCL 5	30		
06	61	x	31		
07	41	–	32		
08	74	R/S	33		
09	24 03	RCL 3	34		
10	24 04	RCL 4	35		
11	61	x	36		
12	24 01	RCL 1	37		
13	24 06	RCL 6	38		
14	61	x	39		
15	41	–	40		
16	74	R/S	41		
17	24 01	RCL 1	42		
18	24 05	RCL 5	43		
19	61	x	44		
20	24 02	RCL 2	45		
21	24 04	RCL 4	46		
22	61	x	47		
23	41	–	48		
24	13 00	GTO 00	49		

REGISTRES
R_0
R_1 a_1
R_2 a_2
R_3 a_3
R_4 b_1
R_5 b_2
R_6 b_3
R_7

Exemple:

Soit A = (2, 5, 2)
B = (3, 3, −4)

Solution:

A × B = (−26, 14, −9)

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre A en mémoire	a_1	STO	1			
		a_2	STO	2			
		a_3	STO	3			
3	Mettre B en mémoire	b_1	STO	4			
		b_2	STO	5			
		b_3	STO	6			
4	Calcul du produit vectoriel		f	PRGM	R/S		c_1
			R/S				c_2
			R/S				c_3
5	Pour un nouveau cas, aller en 2.						

MODULE, PRODUIT SCALAIRE ET ANGLE DE DEUX VECTEURS

Soit deux vecteurs $\vec{a} = (a_1, a_2, ..., a_n)$ et $\vec{b} = (b_1, b_2, ..., b_n)$.
Le module de $\vec{a}$ ($|\vec{a}|$) se calcule au moyen de la formule suivante:

$$|\vec{a}| = \sqrt{a_1^2 + a_2^2 + ... + a_n^2}$$

similairement,

$$|\vec{b}| = \sqrt{b_1^2 + b_2^2 + ... + b_n^2}$$

Le produit scalaire de $\vec{a}$ et $\vec{b}$ ($\vec{a} \cdot \vec{b}$) se calcule au moyen de la formule suivante:

$\vec{a} \cdot \vec{b} = a_1 b_1 + a_2 b_2 + ... + a_n b_n$

L'angle θ des vecteurs $\vec{a}$ et $\vec{b}$ se calcule au moyen de la formule suivante:

$$\theta = \arccos \left(\frac{\vec{a} \cdot \vec{b}}{|\vec{a}| \cdot |\vec{b}|} \right)$$

Cet angle se calcule dans n'importe quel mode angulaire (mode DEGRÉS: degrés décimaux).

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	31	↑
02	14 02	g x^2
03	23 51 01	STO + 1
04	22	R↓
05	21	x⇄y
06	31	↑
07	14 02	g x^2
08	23 51 00	STO + 0
09	22	R↓
10	61	x
11	23 51 02	STO + 2
12	13 00	GTO 00
13	24 02	RCL 2
14	24 00	RCL 0
15	24 01	RCL 1
16	61	x
17	14 02	f $\sqrt{x}$
18	71	÷
19	15 05	g COS^{-1}
20	13 00	GTO 00
21		
22		
23		
24		

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 Σa_i^2
R_1 Σb_i^2
R_2 $\Sigma a_i b_i$
R_3
R_4
R_5
R_6
R_7

Exemple:

Soit A = (2, 5, 2)
B = (3, 3, −4).

Solution:

$|\vec{a}| = 5.74$
$|\vec{b}| = 5.83$
$\vec{a} \cdot \vec{b} = 13.00$
$\theta = 67.16°$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	REG	f	PRGM	
3	Effectuer 3 pour i = 1, ... n						
	Introduire a_i et b_i	a_i	↑				
		b_i	R/S				
4	Calcul du module de $\vec{a}$		RCL	0	f	$\sqrt{x}$	$\lvert\vec{a}\rvert$
5	Calcul du module de $\vec{b}$		RCL	1	f	$\sqrt{x}$	$\lvert\vec{b}\rvert$
6	Calcul de $\lvert\vec{a}\rvert \cdot \lvert\vec{b}\rvert$		RCL	2			$\lvert\vec{a} \cdot \vec{b}\rvert$
7	Calcul de l'angle de $\vec{a}$ et $\vec{b}$		GTO	13	R/S		θ

SYSTÈME DE 2 ÉQUATIONS À 2 INCONNUES

Soit un système de deux équations à deux inconnues:

$$\begin{cases} ax + by = e \\ cx + dy = f. \end{cases}$$

La méthode de Cramer permet de trouver la solution.

$$x = \frac{\begin{vmatrix} e & b \\ f & d \end{vmatrix}}{\begin{vmatrix} a & b \\ c & d \end{vmatrix}} = \frac{ed - bf}{ad - bc} \qquad y = \frac{\begin{vmatrix} a & e \\ c & f \end{vmatrix}}{\begin{vmatrix} a & b \\ c & d \end{vmatrix}} = \frac{af - ec}{ad - bc}$$

Si ad − bc = 0, le calculateur affiche zéro. Dans ce cas, il existe aucune ou plusieurs solutions.

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	24 03	RCL 3
02	24 05	RCL 5
03	61	x
04	24 02	RCL 2
05	24 06	RCL 6
06	61	x
07	41	−
08	24 01	RCL 1
09	24 05	RCL 5
10	61	x
11	24 02	RCL 2
12	24 04	RCL 4
13	61	x
14	41	−
15	23 00	STO 0
16	71	÷
17	74	R/S
18	24 01	RCL 1
19	24 06	RCL 6
20	61	x
21	24 03	RCL 3
22	24 04	RCL 4
23	61	x
24	41	−

AFFICHAGE		TOUCHES
PAS	CODE	
25	24 00	RCL 0
26	71	÷
27	13 00	GTO 00
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 ad − bc
R_1 a
R_2 b
R_3 e
R_4 c
R_5 d
R_6 f
R_7

Exemple:

$5x-3y=12$
$2x+y=9$

Solution:

$x=3.55$
$y=1.91$

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire les	a	STO	1			
	coefficients	b	STO	2			
		e	STO	3			
		c	STO	4			
		d	STO	5			
		f	STO	6			
3	Calcul de x et de y		f	PRGM	R/S		x
			R/S				y
4	Pour un nouveau cas, aller en 2.						

CHAPITRE 2: CALCULS FINANCIERS

De nombreux programmes financiers ayant des caractéristiques communes, nous pensons qu'il est intéressant de dire un mot des paramètres et des termes utilisés dans les programmes qui suivent.

Les principaux paramètres rencontrés dans les problèmes financiers sont les suivants:

- n: nombre de périodes
- i: taux d'intérêt périodique exprimé sous forme décimale. Un taux annuel d'intérêt de 6% sera exprimé par 0.06, le taux mensuel proportionnel valant $\frac{0.06}{12}=0.005$.
- PMT: montant d'un versement périodique
- PV: valeur actuelle (au début de la première période)
- FV: valeur future (à la fin de la dernière période)

AMORTISSEMENT D'UN EMPRUNT INTÉRÊTS CUMULÉS – CAPITAL RESTANT DÛ



Un des problèmes financiers les plus courants est l'établissement du tableau d'amortissement d'un emprunt remboursé par annuités constantes de fin de période. Chaque versement périodique se décompose en effet en une part d'intérêt payé et une part de capital remboursé (ou amorti).

Une personne, qui a emprunté par exemple 150 000F sur 30 ans à un taux d'intérêt annuel de 8%, effectue un premier remboursement mensuel de 1100.65F. La première réaction consiste à retrancher la totalité de ce versement de la dette pour obtenir la dette résiduelle, ce qui est loin d'être un bon raisonnement. En effet, sur les 1100.65F, seuls 100.65F de capital ont été remboursés pour 1000F d'intérêts payés! Le principe est le suivant: les intérêts payés pour un versement donné sont proportionnels au montant du capital restant à rembourser (le coefficient de proportionnalité est, bien sûr, le taux périodique d'intérêt); l'amortissement du capital pour cette période est la différence entre le versement mensuel et les intérêts calculés.

Ce programme vous permet de calculer le montant des intérêts versés pour un ou plusieurs versements, ainsi que le montant du capital restant à rembourser. Introduire d'abord le montant du prêt, le taux d'intérêt périodique, le montant de chaque remboursement, puis les numéros du premier (J) et du dernier (K) remboursement de la période considérée. Le programme calcule le montant des intérêts cumulés entre les remboursements J et K inclus et le capital restant dû après le $K^{\text{ième}}$ remboursement. Si vous désirez connaître le montant des intérêts payés pour un versement déterminé, il vous suffit de faire $K = J$. Ce programme peut aussi être utilisé pour dresser un tableau d'amortissement indiquant le capital restant dû après plusieurs remboursements successifs; pour cela, faire $J = 1$ et augmenter K de 1 à chaque itération. Le HP-25 donne le montant total des intérêts payés pour les K premiers remboursements et le capital restant dû après le $K^{\text{ième}}$ remboursement.

Formules:

$$BAL_K = \frac{1}{(1+i)^{-K}} \left[PMT \, \frac{(1+i)^{-K}-1}{i} + PV \right]$$

$$Int_{J-K} = BAL_K - BAL_{J-1} + (K-J+1)\,PMT$$

où BAL_n: capital restant dû après le $n^{\text{ième}}$ remboursement
Int_{J-K}: montant des intérêts versés pour les remboursements J à K
PV: montant de l'emprunt
PMT: montant d'un remboursement
i: taux d'intérêt périodique

Remarques:

1. Le taux d'intérêt périodique i doit être introduit sous forme décimale. Par exemple, pour rembourser par mensualités un emprunt de taux d'intérêt annuel 9%, le taux d'intérêt mensuel à introduire est $i = \frac{0.09}{12} = 0.0075$
2. Ce programme est utilisable pour tout emprunt amorti par remboursement constant.

Remarques sur la programmation:

Dans de nombreux programmes financiers, les expressions $(1+i)$ et $(1+i)^n$ sont utilisées plusieurs fois dans le même programme. Il est préférable de les calculer une seule fois et de les mettre en mémoire. Dans ce programme, les valeurs de $(1+i)^{-K}$ et $(1+i)^{-J}$ sont calculés une seule fois, puis mises en mémoire dans le registre R_7; vous économisez ainsi des pas de programme et du temps d'exécution.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00							
01	24 01	RCL 1	i				Calcul de BAL_K
02	01	1	1	i			
03	51	+	1 + i				
04	24 05	RCL 5	K	1 + i			
05	32	CHS	−K	1 + i			
06	14 03	f y^x	$(1+i)^{-K}$				
07	23 07	STO 7	$(1+i)^{-K}$				
08	01	1	1	$(1+i)^{-K}$			
09	41	−	$(1+i)^{-K}-1$				
10	24 01	RCL 1	i	$(1+i)^{-K}-1$			
11	71	÷	s				$s = [(1+j)^{-k}-1] \div i$
12	24 02	RCL 2	PMT	s			
13	61	x	PMT s				
14	24 03	RCL 3	PV	PMT s			
15	51	+	PMT s + PV				
16	24 07	RCL 7	$(1+i)^{-K}$	PMT s + PV			
17	71	÷	BAL_K				
18	23 06	STO 6	BAL_K				
19	24 01	RCL 1	i	BAL_K			R_6 — BAL_K
20	01	1	1	i	BAL_K		Calcul de BAL_{J-1}
21	51	+	(1 + i)	BAL_K			
22	24 04	RCL 4	J	(1 + i)	BAL_K		
23	01	1	1	J	(1 + i)	BAL_K	
24	41	−	J − 1	(1 + i)	BAL_K	BAL_K	
25	32	CHS	− (J − 1)	(1 + i)	BAL_K	BAL_K	
26	14 03	f y^x	$(1+i)^{-(J-1)}$	BAL_K	BAL_K	BAL_K	
27	23 07	STO 7	$(1+i)^{1-J}$	BAL_K	BAL_K	BAL_K	
28	01	1	1	$(1+i)^{1-J}$	BAL_K	BAL_K	
29	41	−	$(1+i)^{1-J}-1$	BAL_K	BAL_K	BAL_K	
30	24 01	RCL 1	i	$(1+i)^{1-J}-1$	BAL_K	BAL_K	
31	71	÷	s	BAL_K	BAL_K	BAL_K	$s = [(1+j)^{1-J}-1] \div i$
32	24 02	RCL 2	PMT	s	BAL_K	BAL_K	
33	61	x	PMT s	BAL_K	BAL_K	BAL_K	
34	24 03	RCL 3	PV	PMT s	BAL_K	BAL_K	
35	51	+	PMT s + PV	BAL_K	BAL_K	BAL_K	
36	24 07	RCL 7	$(1+i)^{1-J}$	PMT s + PV	BAL_K	BAL_K	
37	71	÷	BAL_{J-1}	BAL_K	BAL_K	BAL_K	
38	41	−	Diff	BAL_K	BAL_K	BAL_K	$Diff = BAL_K - BAL_{J-1}$
39	24 05	RCL 5	K	Diff	BAL_K	BAL_K	K − J + 1: nombre de
40	24 04	RCL 4	J	K	Diff	BAL_K	versements entre J et K
41	41	−	K−J	Diff	BAL_K	BAL_K	
42	01	1	1	K−J	Diff	BAL_K	
43	51	+	K − J + 1	Diff	BAL_K	BAL_K	
44	24 02	RCL 2	PMT	m	Diff	BAL_K	m = K − J + 1
45	61	x	m PMT	Diff	BAL_K	BAL_K	m PMT est payé
46	51	+	Int_{J-K}	BAL_K	BAL_K	BAL_K	Affichage de Int_{J-K}
47	74	R/S	Int_{J-K}	BAL_K	BAL_K	BAL_K	
48	21	$x \rightleftarrows y$	BAL_K	Int_{J-K}	BAL_K	BAL_K	Affichage de BAL_K
49	13 00	GTO 00	BAL_K	Int_{J-K}	BAL_K	BAL_K	

REGISTRES	
R_0	
R_1	i
R_2	PMT
R_3	PV
R_4	J
R_5	K
R_6	BAL_K
R_7	$(1+i)^{-n}$

Exemple:

Une hypothèque est telle que le premier versement a lieu à la fin du mois d'octobre 1974 (c'est-à-dire qu'octobre est la première période de paiement). Il s'agit d'un prêt de 25000F à 8% et les paiements mensuels sont de 200F. Quels sont les intérêts versés en 1974 (périodes 1 à 3) et 1975 (périodes 4 à 15) et quel est le montant du capital restant dû à la fin de chacune de ces années? Dresser également un tableau donnant les intérêts accumulés et les capitaux restant dus pour les 5 premières années de l'hypothèque (périodes 12, 24, 36, 48 et 60).

Solution:

(Introduire le taux mensuel i sous forme décimale)

.08 [↑] 12 [÷] [STO] [1] 200 [STO] [2] 25000 [STO] [3] 1
[STO] [4] 3 [STO] [5] [f] [PRGM] [R/S] ——→ 499.33
(intérêts versés en 1974)

[R/S] ——→ 24899.33
(capital restant dû fin 1974)

4 [STO] [4] 15 [STO] [5] [R/S] ——→ 1976.65
(intérêts versés en 1975)

[R/S] ——→ 24475.98
(capital restant dû fin 1975)

Puis, dressez le tableau d'amortissement:

1 [STO] [4] 12 [STO] [5] [R/S] ——→ 1985.00
(intérêts première année)

[R/S] ——→ 24585.00
(capital restant dû après la première année)

24 [STO] [5] [R/S] ——→ 3935.56
(intérêts deuxième année)

[R/S] ——→ 24135.56
(capital restant dû après la deuxième année)

36 [STO] [5] [R/S] ——→ 5848.81
(intérêts troisième année)

[R/S] ——→ 23648.81
(capital restant dû après la troisième année)

48 [STO] [5] [R/S] ——→ 7721.67
(intérêts quatrième année)

[R/S] ——→ 23121.67
(capital restant dû après la quatrième année)

60 [STO] [5] [R/S] ——→ 9550.77
(intérêts cinquième année)

[R/S] ——→ 22550.77
(capital restant dû après la cinquième année)



N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire les données suivantes:						
	Taux d'intérêt périodique (décimale)	i	STO	1			
	Montant d'un remboursement	PMT	STO	2			
	Montant du prêt	PV	STO	3			
	Numéro de la période de départ	J	STO	4			
	Numéro de la période finale	K	STO	5	f	PRGM	
3	Calcul du montant des intérêts						
	versés pendant les périodes J-K		R/S				Int_{J-K}
4	Affichage du capital restant dû						
	après le Kième remboursement		R/S				BAL_K
5	Pour modifier la valeur d'une donnée, mettre en mémoire la						
	nouvelle valeur dans le registre correspondant et aller en 3.						

EMPRUNT: MONTANT, NOMBRE DE REMBOURSEMENTS ET MONTANT D'UN REMBOURSEMENT (VERSEMENTS À TERME ÉCHU)



Ce programme calcule le montant d'un emprunt à annuités constantes (PV), le nombre de remboursement (n) ou le montant d'un remboursement (PMT), connaissant deux de ces trois données et le taux d'intérêt.

Le taux d'intérêt périodique i doit être exprimé sous forme décimale (exemple 6% : 0.06).

Les formules utilisées sont les suivantes:

$$PMT = PV \left[\frac{i}{1-(1+i)^{-n}} \right] \qquad PV = PMT \left[\frac{1-(1+i)^{-n}}{i} \right]$$

$$n = -\frac{\ln(1 - i\,PV/PMT)}{\ln(1+i)}$$

Remarque:

Les versements sont effectués à la fin de chaque période (à terme échu).

AFFICHAGE PAS	CODE	TOUCHES
00		
01	01	1
02	24 02	RCL 2
03	01	1
04	51	+
05	24 01	RCL 1
06	32	CHS
07	14 03	f y^x
08	41	–
09	24 02	RCL 2
10	21	x⇄y
11	71	÷
12	24 04	RCL 4
13	61	x
14	13 00	GTO 00
15	01	1
16	24 02	RCL 2
17	01	1
18	51	+
19	24 01	RCL 1
20	32	CHS
21	14 03	f y^x
22	41	–
23	24 02	RCL 2
24	71	÷

AFFICHAGE PAS	CODE	TOUCHES
25	24 03	RCL 3
26	61	x
27	13 00	GTO 00
28	01	1
29	24 04	RCL 4
30	24 03	RCL 3
31	71	÷
32	24 02	RCL 2
33	61	x
34	41	–
35	14 07	f LN
36	24 02	RCL 2
37	01	1
38	51	+
39	14 07	f LN
40	71	÷
41	32	CHS
42	13 00	GTO 00
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1 n
R_2 i
R_3 PMT
R_4 PV
R_5
R_6
R_7

Exemples:

1. Un particulier emprunte 15 000F à un taux annuel de 9.5% (0.095); il compte rembourser cette somme sur 36 mois. Quel sera le montant des mensualités?
2. De quelle somme pouvez-vous disposer, si vous désirez acquitter des mensualités de 750F pendant 30 mois à un taux de 9.5%?
3. Vous empruntez 20 000F à un taux d'intérêt annuel de 9.5%. A raison de 1000F par mois, combien de temps vous faudra-t-il pour rembourser cette somme?

Solutions:

Pour obtenir le taux d'intérêt périodique exprimé sous forme décimale, diviser 0.095 par 12.

1. 480.49F
2. 19 957.77F
3. 21.86 mois

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Pour le montant d'un remboursement	n	STO	1			
		i	STO	2			
		PV	STO	4			
			f	PRGM	R/S		PMT
3	Pour le montant de l'emprunt	n	STO	1			
		i	STO	2			
		PMT	STO	3			
			GTO	15	R/S		PV
4	Pour le nombre de remboursements	i	STO	2			
		PMT	STO	3			
		PV	STO	4			
			GTO	28	R/S		n
5	Pour un nouveau cas, aller en 2, 3 ou 4						

TAUX D'INTÉRÊT D'UN EMPRUNT (VERSEMENTS DE FIN DE PÉRIODE)



Ce programme calcule le taux d'intérêt d'un emprunt à annuités constantes versées en fin de chaque période, connaissant le nombre de périodes (n), la valeur actuelle ou le montant initial de l'emprunt (PV) et le montant d'un remboursement (PMT).

Ce programme calcule le taux périodique par la méthode d'itération de Newton:

$$i_{k+1} = i_k - \frac{f(i_k)}{f'(i_k)}$$

où:

$$f(i) = \frac{1-(1+i)^{-n}}{i} - \frac{PV}{PMT}$$

La valeur initiale du taux est donnée par:

$$i_o = \frac{PMT}{PV} - \frac{PV}{n^2\ PMT}$$

AFFICHAGE PAS	CODE	TOUCHES
00		
01	24 03	RCL 3
02	31	↑
03	15 22	g 1/x
04	21	x⇄y
05	24 01	RCL 1
06	15 02	g x^2
07	71	÷
08	41	–
09	23 02	STO 2
10	24 03	RCL 3
11	24 02	RCL 2
12	61	x
13	01	1
14	24 02	RCL 2
15	01	1
16	51	+
17	24 01	RCL 1
18	32	CHS
19	14 03	f y^x
20	23 05	STO 5
21	41	–
22	41	–
23	24 01	RCL 1
24	24 02	RCL 2

AFFICHAGE PAS	CODE	TOUCHES
25	15 22	g 1/x
26	01	1
27	51	+
28	71	÷
29	01	1
30	51	+
31	24 05	RCL 5
32	61	x
33	01	1
34	41	–
35	24 02	RCL 2
36	71	÷
37	71	÷
38	23 51 02	STO + 2
39	15 03	g ABS
40	33	EEX
41	06	6
42	32	CHS
43	14 41	f x<y
44	13 10	GTO 10
45	24 02	RCL 2
46	13 00	GTO 00
47		
48		
49		

REGISTRES
R_0
R_1 n
R_2 i
R_3 PV/PMT
R_4 $(1+i)^{-n}$
R_5
R_6
R_7

Exemple:

Vous prenez un crédit de 15000F en vue d'acheter une voiture. Vous le rembourserez par 36 mensualités de 500F. Quel est le taux du crédit?

Solution:

1.02% par mois, soit environ 12.55% par an.

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire le nombre remboursements	n	STO	1			
3	Introduire le montant						
	du crédit et celui de chaque	PV	↑				
	remboursement	PMT	÷	STO	3		PV/PMT
4	Calcul du taux d'intérêt		f	PRGM	R/S		i (décimale)
			EEX	2	x		i (%)
5	Pour un nouveau cas, aller en 2.						

INTÉRÊTS COMPOSÉS CAPITALISATION, ACTUALISATION



Ce programme s'applique à un capital unique placé à intérêts composés. Les paramètres sont le nombre de périodes n, le taux d'intérêt périodique i, la valeur actuelle du capital PV, la valeur future du capital FV et le montant des intérêts acquis I. On peut obtenir n'importe quel paramètre à partir des autres.

Les formules utilisées sont les suivantes:

$$n = \frac{\ln (FV/PV)}{\ln (1+i)} \qquad i = \left(\frac{FV}{PV}\right)^{1/n} - 1 \qquad PV = FV\,(1+i)^{-n}$$

$$FV = PV\,(1+i)^n \qquad I = PV\,[(1+i)^n - 1]$$

AFFICHAGE PAS	CODE	TOUCHES
00		
01	24 05	RCL 5
02	24 04	RCL 4
03	71	÷
04	14 07	f LN
05	24 02	RCL 2
06	01	1
07	51	+
08	14 07	f LN
09	71	÷
10	13 00	GTO 00
11	24 05	RCL 5
12	24 04	RCL 4
13	71	÷
14	24 01	RCL 1
15	15 22	g 1/x
16	14 03	f y^x
17	01	1
18	41	–
19	13 00	GTO 00
20	24 02	RCL 2
21	01	1
22	51	+
23	24 01	RCL 1
24	32	CHS

AFFICHAGE PAS	CODE	TOUCHES
25	14 03	f y^x
26	24 05	RCL 5
27	61	x
28	13 00	GTO 00
29	24 02	RCL 2
30	01	1
31	51	+
32	24 01	RCL 1
33	14 03	f y^x
34	24 04	RCL 4
35	61	x
36	13 00	GTO 00
37	24 02	RCL 2
38	01	1
39	51	+
40	24 01	RCL 1
41	14 03	f y^x
42	01	1
43	41	–
44	24 04	RCL 4
45	61	x
46	13 00	GTO 00
47		
48		
49		

REGISTRES
R_0
R_1 n
R_2 i
R_3
R_4 PV
R_5 FV
R_6
R_7

Exemples:

1. En supposant que le taux annuel d'inflation est de 10%, en combien de temps les prix doubleront-ils? (PV = 1, FV = 2).
2. A quel taux trimestriel faut-il placer une somme de 1000F pour disposer de 1500F dans 5 ans?
3. Combien vous faut-il investir maintenant au taux d'intérêt de 5.75%, les intérêts étant capitalisés trimestriellement, pour disposer de 3000F dans 5 ans?
4. Quelle est la valeur actuelle acquise par 2000F placés à 5.75% (0.0575) pendant 4 ans, les intérêts étant capitalisés trimestriellement?
5. Quel est le montant des intérêts sur un capital de 1500F placés à 5.5% pendant 10 ans, les intérêts étant capitalisés annuellement?

Solutions:

1. 7.27 ans
2. 0.0205 (taux trimestriel) = 8.19% (taux annuel)
3. 2255.02F (i = 0.0575/4)
4. 2513.08F (i = 0.0575/4)
5. 1062.22F (i = 0.055)

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Calcul du nombre de périodes	i (décimale)	STO	2			
		PV	STO	4			
		FV	STO	5			
			f	PRGM	R/S		n
3	Calcul du taux d'intérêt						
	périodique	n	STO	1			
		PV	STO	4			
		FV	STO	5			
			GTO	11	R/S		i (décimale)
4	Calcul de la valeur actuelle	n	STO	1			
		i (décimale)	STO	2			
		FV	STO	5			
			GTO	20	R/S		PV
5	Calcul de la valeur future	n	STO	1			
		i (décimale)	STO	2			
		PV	STO	4			
			GTO	29	R/S		FV
6	Calcul du montant	n	STO	1			
	des intérêts	i (décimale)	STO	2			
		PV	STO	4			
			GTO	37	R/S		I
7	Pour un nouveau cas,						
	aller en 2, 3, 4, 5 ou 6						

PLAN D'ÉPARGNE
MONTANT D'UN VERSEMENT, VALEUR FUTURE, NOMBRE DE VERSEMENTS



Ce programme calcule le montant d'un versement, la valeur future ou le nombre de versements d'un plan d'épargne, connaissant deux de ces trois données ainsi que le taux périodique d'intérêt.

Soit:
n: nombre de versements
i: taux d'intérêt périodique exprimé sous forme décimale (ex 6% = 0.06)
PMT: montant d'un versement
FV: valeur future
n, PMT ou FV peuvent être calculés à partir des formules suivantes:

$$n = \frac{\ln\left[\frac{FV\,i}{PMT} + (1+i)\right]}{\ln(1+i)} - 1 \qquad PMT = \frac{FV\,i}{(1+i)^{n+1} - (1+i)}$$

$$FV = \frac{PMT}{i}\left[(1+i)^{n+1} - (1+i)\right]$$

Remarque:

Les versements sont effectués en début de chaque période (annuités par terme à échoir).

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	24 02	RCL 2
02	24 05	RCL 5
03	61	x
04	24 03	RCL 3
05	71	÷
06	24 02	RCL 2
07	01	1
08	51	+
09	23 00	STO 0
10	51	+
11	14 07	f LN
12	24 00	RCL 0
13	14 07	f LN
14	71	÷
15	01	1
16	41	–
17	13 00	GTO 00
18	24 05	RCL 5
19	24 02	RCL 2
20	61	x
21	24 02	RCL 2
22	01	1
23	51	+
24	71	÷
25	14 73	f LASTx
26	24 01	RCL 1
27	14 03	f y^x
28	01	1
29	41	–
30	71	÷
31	13 00	GTO 00
32	24 03	RCL 3
33	24 02	RCL 2
34	01	1
35	51	+
36	61	x
37	14 73	f LASTx
38	24 01	RCL 1
39	14 03	f y^x
40	01	1
41	41	–
42	61	x
43	24 02	RCL 2
44	71	÷
45	13 00	GTO 00
46		
47		
48		
49		

REGISTRES
R_0 (1 + i)
R_1 n
R_2 i
R_3 PMT
R_4
R_5 FV
R_6
R_7

Exemples:

1. Vous déposez en début de chaque trimestre 2000F sur un compte d'épargne logement à 4.5% d'intérêt annuel. Combien de versements devez-vous effectuer pour capitaliser 30 000F?
2. Vous désirez accumuler 40 000F en 7 ans. Quel doit être le montant de votre versement mensuel si le taux d'intérêt annuel est de 6.5%?
3. De quelle somme disposerez-vous dans 3 ans en déposant 500F à chaque début de mois sur un compte d'épargne à 4.5% par an?

Solutions:

1. 13.79 trimestres (3,45 années)
2. 375.28F
3. 19305.17F

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Calcul du nombre de						
	versements	i (décimale)	STO	2			
		PMT	STO	3			
		FV	STO	5			
			f	PRGM	R/S		n
3	Calcul du montant						
	d'un versement	n	STO	1			
		i (décimale)	STO	2			
		FV	STO	5			
			GTO	18	R/S		PMT
4	Calcul de la valeur future	n	STO	1			
		i (décimale)	STO	2			
		PMT	STO	3			
			GTO	32	R/S		FV
5	Pour un nouveau cas,						
	aller en 2, 3 ou 4.						

RENTABILITÉ D'UN INVESTISSEMENT PAR LA MÉTHODE DES FLUX ACTUALISÉS: VALEUR ACTUELLE NETTE, TAUX INTERNE DE RENTABILITÉ

Ce programme détermine la valeur actuelle nette d'une série de flux de trésorerie (cash-flows), ce qui permet de savoir si un investissement a été rentable à un taux donné. On connaît le montant de l'investissement V_0 ainsi que les bénéfices ou flux nets réalisés pour les n périodes envisagées C_1, C_2, ..., C_n. On se fixe un taux de rentabilité i (périodique et décimal) et le calcul consiste à actualiser chacun des bénéfices à l'époque 0, à faire la somme de ces bénéfices actualisés et la balance avec l'investissement initial. Si le résultat NPV_k est positif, l'investissement a été rentable à i% (le taux est donc plus élevé que i%). Si le résultat est négatif, l'investissement n'a pas été rentable au taux i% espéré.

$$NPV_k = -V_0 + \sum_{j=1}^{k} \frac{C_j}{(1+i)^j}$$

Ce programme permet de déterminer le taux i% réellement réalisé, par approximations successives. Le principe est le suivant: au taux cherché i, la valeur actuelle nette NPV doit s'annuler. On peut converger très rapidement vers la solution en utilisant une interpolation linéaire des taux.



i	i_1	i_2 ...	i_n
NPV	NPV_1	NPV_2 ...	$NPV_n \sim 0$

Cette interpolation porte sur les deux derniers taux calculés. Il faut se fixer deux taux i_1 et i_2 au départ (si possible proches de la solution), puis calculer les NPV_1 et NPV_2 correspondants. Le taux suivant i_3 sera alors donné par:

$$i_3 = \frac{i_1 \times NPV_2 - i_2 \times NPV_1}{NPV_2 - NPV_1}$$

Calculer de même i_n à partir des points P_2 et P_3 et ainsi de suite jusqu'à i_n tel que NPV_n soit très proche de 0.

Remarque:

Il serait intéressant d'ajouter le calcul d'interpolation dans le programme.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	24 01	RCL 1
02	01	1
03	23 04	STO 4
04	51	+
05	23 02	STO 2
06	71	÷
07	24 00	RCL 0
08	41	−
09	24 04	RCL 4
10	14 74	f PAUSE
11	21	x⇄y
12	23 03	STO 3
13	74	R/S
14	24 02	RCL 2
15	24 04	RCL 4
16	01	1
17	51	+
18	23 04	STO 4
19	14 03	f y^x
20	71	÷
21	24 03	RCL 3
22	51	+
23	13 09	GTO 09
24		

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 V_0
R_1 i
R_2 (1 + i)
R_3 NPV_k
R_4 k
R_5
R_6
R_7

Exemple:

Vous avez la possibilité d'investir 150 000F dans un certain projet. A partir des bénéfices réels suivants, et moyennant un taux d'actualisation de 10%, cet investissement est-il rentable?

Année	Cash-flow
1	30 000F
2	26 300F
3	50 000F
4	55 600F
5	45 200F

Solution:

(introduire i sous forme décimale 0.10)

$NPV_1 = -122\,727.27F$
$NPV_2 = -100\,991.74F$
$NPV_3 = -63\,426.00F$
$NPV_4 = -25\,450.45F$
$NPV_5 = 2\,615.20F$

La valeur actuelle nette C_5 étant positive, l'affaire est rentable. On pourrait calculer le taux exact en essayant d'autres taux.

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire l'investisse-						
	ment initial et le taux	V_0	STO	0			
	d'actualisation	i (décimale)	STO	1	f	PRGM	
3	Effectuer 3 pour k=1, ..., n:						
	Introduire C_k et	C_k	R/S				(k)
	calcul de NPV_k						NPV_k
4	Pour un nouveau cas, aller en 2.						

CALENDRIER : JOUR DE LA SEMAINE, NOMBRE DE JOURS ENTRE DEUX DATES

Ce programme calcule le jour de la semaine pour une date donnée et le nombre de jours exact entre deux dates, comprises entre le 1er mars 1700 (jour 1) et le 28 février 2100. Les jours d'une semaine sont affichés par un numéro :

0 : dimanche
1 : lundi
2 : mardi, etc.

Le numéro du jour N se calcule d'après la formule suivante (a : année, m : mois, j : jour) :

$$N = [365{,}25\, g(a, m)] + [30{,}6\, f(m)] + J - 621049$$

où

$$g(a, m) = \begin{cases} a-1 \text{ si } m=1 \text{ ou } 2 \\ a \text{ si } m>2 \end{cases} \quad \text{et } f(m) = \begin{cases} m+13 \text{ si } m=1 \text{ ou } 2 \\ m+1 \text{ si } m>2 \end{cases}$$

(m) représente la partie entière d'un nombre ; ainsi [6.34] = 6.

Remarque :

Pour les jours compris entre le 1er mars 1700 et le 28 février 1800, il faut ajouter 2 jours à la solution et un jour pour ceux compris entre le 1er mars 1800 et le 28 février 1900.

AFFICHAGE PAS	CODE	TOUCHES
00		
01	03	3
02	24 01	RCL 1
03	14 41	f x<y
04	13 09	GTO 09
05	01	1
06	51	+
07	24 03	RCL 3
08	13 15	GTO 15
09	01	1
10	03	3
11	51	+
12	24 03	RCL 3
13	01	1
14	41	–
15	03	3
16	06	6
17	05	5
18	73	·
19	02	2
20	05	5
21	61	x
22	14 01	f INT
23	21	x⇄y
24	03	3

AFFICHAGE PAS	CODE	TOUCHES
25	00	0
26	73	·
27	06	6
28	61	x
29	14 01	f INT
30	51	+
31	24 02	RCL 2
32	51	+
33	06	6
34	02	2
35	01	1
36	00	0
37	04	4
38	09	9
39	41	–
40	74	R/S
41	07	7
42	71	÷
43	15 01	g FRAC
44	07	7
45	61	x
46	13 00	GTO 00
47		
48		
49		

REGISTRES	
R_0	
R_1	Mois
R_2	Jour
R_3	Année
R_4	
R_5	
R_6	
R_7	Temporaire

Exemples:

1. Quel jour de la semaine était le 4 juillet 1776?
2. Combien de jours se sont écoulés entre le 27 mars 1948 et le 7 avril 1975?

Solutions:

1. Jeudi (4) (penser à ajouter 2 jours)
2. 9872 jours

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire le mois	m	STO	1			
	le jour	j	STO	2			
	l'année	a	STO	3			
3	Calcul de N (m, j, a)		f	PRGM	R/S		N(m, j, a)
4	Pour le jour de la semaine, aller en 8						
5	Pour le calcul de jours entre 2 dates, mettre d'abord en						
	mémoire N		STO	7			
6	Répéter les opérations 2 et 3						
	pour la deuxième date		RCL	7	–		# jours
7	Pour un nouveau cas, aller en 2						
8	Pour le jour de la semaine (0=dimanche)		R/S				jour (0, ..., 6)
9	Pour un nouveau cas, aller en 2.						

CHAPITRE 3: JEUX

SIMULATION D'UN ALUNISSAGE

Imaginez un instant les difficultés d'un alunissage avec réserves limitées de carburant: il s'agit de poser un engin, en douceur, sur le sol lunaire. L'allumage des rétrofusées permet de freiner la descente, mais le carburant ne doit pas être brûlé trop vite ou trop tôt, car vous risqueriez de vous trouver à 30 mètres du sol, les réservoirs à sec, avec toutes les conséquences fâcheuses que cela entraînerait! La bonne manœuvre consiste, bien sûr, à doser et à espacer les coups de freins, de manière à toucher le sol lunaire à une vitesse très faible.

Le jeu démarre alors que l'engin est, à 500 mètres, à une vitesse de 50 m/s. Vitesse et altitude sont affichées sous la forme –50.500, l'altitude étant à droite du point décimal, et la vitesse à gauche. Le signe (–) indique que le mouvement est descendant. Une vitesse affichée sans partie décimale, par exemple –50, signifie que vous vous êtes écrasés à une vitesse de 50 m/s. En termes de jeu, cela veut dire que vous avez perdu; dans la réalité, la signification serait encore bien moins amusante!

Démarrons le jeu avec 120 litres de carburant. A chaque étape de la descente, vous pouvez brûler autant de carburant que vous voulez, dans la limite des réserves encore disponibles. Il est possible de ne pas brûler de carburant. Brûler 5 litres annule la gravitation lunaire et permet de garder une vitesse constante. Brûler plus de 5 litres modifie la vitesse vers le haut. Vous devez faire attention, bien sûr, de ne pas brûler plus de carburant qu'il n'en reste. Si cela se produit, ce sera la chute libre vers un tragique destin! La vitesse finale affichée sera votre vitesse d'impact (généralement très élevée). Vous pouvez afficher à chaque instant votre réserve de carburant en appuyant sur les touches RCL 2.

Formules:

Pour ne pas gâcher l'attrait, nous ne rentrerons pas dans les détails, mais soyez assuré que ce programme est basé sur quelques formules classiques de la physique newtonnienne:

$$x = x_0 + v_0 t + \frac{1}{2} at^2 \quad v = v_0 + at \quad v^2 = v_0^2 + 2\,ax$$

où x, v, a et t sont la distance, la vitesse, l'accélération et le temps.

Remarques:

1. Si vous vous écrasez avant d'être à court de carburant, la vitesse d'impact affichée sera la vitesse atteinte avant le dernier usage de carburant, et non la vitesse réelle d'impact.
2. Les valeurs de carburant brûlé doivent être entières. Toute introduction illicite provoquerait une erreur dans l'affichage de V.X.

Remarques sur la programmation:

Une des particularités intéressantes de ce programme est l'affichage combiné (V.X) de la vitesse et de l'altitude, par exemple –15.0150. Ceci est obtenu par stockage de V et X sous leur forme normale (–15,00; 150,00), puis par division de X par 10000 (10^4) avant la combinaison. Une astuce est également utilisée pour déterminer le signe de V et la nécessité d'ajouter ou de retrancher ($X/10^4$) de V. Si, par exemple, V = –15 et X = 150, il faudrait soustraire ($X/10^4$) de V pour obtenir –15.0150. Mais, si V = 10 et X = 8, il faudrait ajouter ($X/10^4$) à V pour obtenir à l'affichage 10.0050.
Un coup d'œil aux pas 2 à 12 du programme vous montrera comment la fonction valeur absolue (g ABS) a été utilisée pour cette astuce.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00							
01	14 11 04	f FIX 4					Affichage à 4 décimales
02	24 00	RCL 0	X				Elaboration de l'affichage V.X
03	33	EEX	1. 00	X			
04	04	4	1. 04	X			
05	71	÷	$X/10^4$				Division par 10^4 pour placer
06	24 01	RCL 1	V	$X/10^4$			X à droite du point décimal
07	15 41	g x<0	V	$X/10^4$			
08	13 11	GTO 11	V	$X/10^4$			
09	51	+	$V + X/10^4$				Affichage de $V+X/10^4$
10	13 13	GTO 13	$V + X/10^4$				
11	21	x⇄y	$X/10^4$	V			
12	41	–	$V - X/10^4$				Affichage de $V-X/10^4$
13	74	R/S	V.X				V.X soit $V \pm (X/10^4)$
14	24 02	RCL 2	F	B			Introduction du carburant B
15	14 41	f x<y	F	B			carburant à brûler > réserves
16	13 34	GTO 34	F	B			Oui. On va s'écraser
17	22	R↓	B			F	Non. Calcul de A, X, V
18	23 41 02	STO – 2	B			F	Nouveau carburant restant
19	05	5	5	B			Gravitation = 5 unités
20	41	–	B – 5				Accélération = B – 5
21	23 03	STO 3	A				
22	02	2	2	A			
23	71	÷	A/2				
24	24 00	RCL 0	X	A/2			
25	51	+	X + A/2				
26	24 01	RCL 1	V	X + A/2			
27	51	+	X + V + A/2				Altitude: X ← X + V + A/2
28	23 00	STO 0	X				
29	15 41	g x<0	X				Altitude < sol?
30	13 44	GTO 44	X				Oui. On s'est écrasé
31	24 03	RCL 3	A	X			Non. Calcul de V
32	23 51 01	STO + 1	A	X			Nouvelle vitesse: V ← V + A
33	13 02	GTO 02	A	X			Afficher V.X
34	24 01	RCL 1	V				Plus de carburant
35	15 02	g x^2	V^2				Afficher la vitesse
36	24 00	RCL 0	X	V^2			$V = (V^2 + 2\,g\,X)^{1/2}$
37	01	1	1	X	V^2		avec g = 5
38	00	0	10	X	V^2		
39	61	×	10 X	V^2			
40	51	+	V^2 + 10 X				
41	14 02	f $\sqrt{x}$	V				
42	32	CHS	V				Vitesse d'impact < 0
43	23 01	STO 1	V				
44	24 01	RCL 1	V				Branchement depuis 30
45	14 11 00	f FIX 0	V				Affichage de V entier
46	13 00	GTO 00	V				pour montrer l'écrasement
47							
48							
49							

REGISTRES	
R_0	X
R_1	V
R_2	Carburant
R_3	Accélération
R_4	
R_5	
R_6	
R_7	

Exemple:

500 STO 0 50 CHS STO 1 120 STO 2

f PRGM R/S ⟶ –50.0500

0 R/S ⟶ –55.0448

5 R/S ⟶ –55.0393

(la vitesse reste constante quand on brûle 5 unités)

30 R/S ⟶ –30.0350

0 R/S ⟶ –35.0318

0 R/S ⟶ –40.0280

0 R/S ⟶ –45.0238

0 R/S ⟶ –50.0190

RCL 2 ⟶ 85.0000

(carburant restant)

f PRGM R/S ⟶ –50.0190

(affichage de V.X)

10 R/S ⟶ –45.0143

0 R/S ⟶ –50.0095

RCL 2 ⟶ 75.0000

10 R/S ⟶ –45.0048

25 R/S ⟶ –25.0013

20 R/S ⟶ –25.

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser	X	500	STO	0		500.00
		V	50	CHS	STO	1	-50.00
		Carburant	120	STO	2		120.00
3	Afficher le V.X initial		f	PRGM	R/S		-50.0500
4	Brûler du carburant	B					
			R/S				V.X
5	Recommencer l'opération 4 jusqu'à l'alunissage en						
	douceur ou non						
6	Pour afficher le						
	carburant restant		RCL	2			Carburant
7	Pour afficher V.X						
			f	PRGM	R/S		V.X
8	Pour un nouveau jeu, aller en 2.						

NIMB

Les règles du jeu de Nimb sont très simples: N objets sont mis en jeu, N étant un nombre entier positif. Chaque joueur retire, à son tour, 1, 2 ou 3 objets, jusqu'à ce qu'il n'en reste plus qu'un. Le joueur qui se trouve obligé de prendre le dernier a perdu.

Au départ, il faut indiquer à la machine le nombre d'objets mis en jeu, c'est-à-dire la valeur de N. Après chaque soustraction, la machine affiche le nombre d'objets restant. Un signe négatif indique que c'est à vous de jouer, un signe positif que c'est au HP-25 de jouer.

En tant que «challenger», c'est à vous de jouer le premier. Il vous est possible de gagner, mais le HP-25 est évidemment un champion à ce jeu, et il ne vous pardonnera aucune erreur. (Mais n'oubliez pas que le HP-25 a en vous une confiance naïve, il attend donc de votre part un jeu loyal: vous ne devez pas soustraire de nombres autres que 1, 2 ou 3.)

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	31	↑
02	01	1
03	23 02	STO 2
04	22	R↓
05	23 41 00	STO – 0
06	24 00	RCL 0
07	15 71	g x=0
08	13 41	GTO 42
09	23 61 02	STO × 2
10	24 02	RCL 2
11	74	R/S
12	21	x⇄y
13	15 51	g x≥0
14	13 16	GTO 17
15	21	x⇄y
16	13 01	GTO 02
17	01	1
18	32	CHS
19	23 02	STO 2
20	00	0
21	23 01	STO 1
22	24 01	RCL 1
23	03	3
24	14 71	f x=y

AFFICHAGE		TOUCHES
PAS	CODE	
25	13 39	GTO 40
26	01	1
27	23 51 01	STO + 1
28	32	CHS
29	24 00	RCL 0
30	51	+
31	24 01	RCL 1
32	41	–
33	04	4
34	71	÷
35	15 01	g FRAC
36	15 61	g x≠0
37	13 21	GTO 22
38	24 01	RCL 1
39	13 04	GTO 05
40	01	1
41	13 04	GTO 05
42	24 02	RCL 2
43	15 41	g x<0
44	13 46	GTO 47
45	24 03	RCL 3
46	13 00	GTO 00
47	24 04	RCL 4
48	14 11 01	f FIX 1
49	13 00	GTO 00

REGISTRES	
R_0	Total
R_1	La machine joue
R_2	±Total
R_3	55178
R_4	3507.1
R_5	
R_6	
R_7	

Exemple:

Jeu avec N = 15

Enlever 3

3 [R/S] ——————————→ 12.

[R/S] ——————————→ -9.

Le HP-25 retranche 3

Enlever 2

2 [R/S] ——————————→ 7.

[R/S] ——————————→ -5.

Le HP-25 retranche 2

Enlever 3

3 [R/S] ——————————→ 2.

[R/S] ——————————→ -1.

Le HP-25 retranche 1

Enlever le dernier

1 [R/S] ——————————→ 55178.

Retourner, le calculateur pour lire, à l'affichage, son commentaire: BLISS («je suis vainqueur»).

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser	55178	STO	3			
		3507.1	STO	4	f	PRGM	
3	Stocker le nombre d'objets						
	(généralement 15) et choisir le	N	STO	0	CHS	f	
	mode d'affichage		FIX	0			-N.
4	Si le nombre affiché est négatif,						
	c'est à vous de jouer	Votre jeu	R/S				+Total
5	Si le nombre affiché est positif,						
	laisser jouer le HP-25		R/S				-Total
6	Recommencer les opérations						
	4 et 5 jusqu'à ce que le jeu soit terminé						
7	A la fin du jeu, lire à l'envers						
	de l'affichage le commentaire du HP-25						
8	Pour un autre jeu, retourner en 3						

UNE LEÇON D'ARITHMÉTIQUE

Hewlett-Packard pense que le calculateur de poche, loin de menacer les principes traditionnels d'un bon enseignement des mathématiques, peut être utilisé, de manière constructive, pour consolider des études dans le domaine de l'arithmétique, l'algèbre, la géométrie, la trigonométrie, le calcul infinitésimal et l'analyse numérique. Ce programme destiné à être utilisé pour l'enseignement, aux enfants, des quatre opérations arithmétiques élémentaires, montre un des nombreux aspects du HP-25 en tant qu'instrument éducatif:

Le principe de ce programme est de poser un problème d'arithmétique et de comparer la réponse correcte à celle que vous donnez. Si votre réponse est juste, le programme continue et un nouveau problème vous est posé. Si elle est fausse, le programme vous pose à nouveau le même problème, vous donnant ainsi une seconde chance.

Pour utiliser le programme, vous devez stocker, dans le registre mémoire R_0, une valeur Max. Cette manipulation a pour but d'empêcher le programme de prendre en considération les nombres aussi grands que la valeur Max. Si vous donnez à Max la valeur 12, par exemple, tout le problème sera traité avec des nombres compris entre 0 et 11. Vous devez en outre stocker dans le registre R_1 un nombre compris entre 0 et 1 qui permettra d'initialiser le générateur de nombres aléatoires donnant les opérandes. Des nombres initiaux différents engendreront des problèmes différents. Si le format d'affichage choisi est ([f] [FIX] [2]), le problème sera affiché de la manière suivante: le premier terme de l'opération sera à gauche du point décimal, le deuxième terme à sa droite. Les nombres 8 et 2, par exemple, seront affichés 8.02. Vous pourrez alors choisir l'opération que vous voulez effectuer: addition $(8+2)$, soustraction $(8-2)$, multiplication (8×2) ou division $(8 \div 2)$. Lorsque vous aurez frappé votre réponse au clavier et relancé l'exécution du programme, celui-ci pourra afficher soit un nouveau problème si votre réponse était juste, soit les deux mêmes nombres sous la forme négative (ce signe négatif indique simplement que la réponse était fausse, et non que les nombres sont négatifs: tous les nombres du problème sont positifs, bien que, évidemment le résultat de certaines soustractions puisse être négatif). Si le problème réapparaît avec un signe négatif, vous devez faire un nouvel essai en proposant une autre réponse. Dès que vous aurez donné la bonne réponse, le programme affichera un nouveau problème.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	24 01	RCL 1
02	15 73	g π
03	15 02	g x^2
04	61	x
05	15 01	g FRAC
06	23 01	STO 1
07	24 00	RCL 0
08	61	x
09	14 01	f INT
10	23 03	STO 3
11	24 01	RCL 1
12	15 73	g π
13	15 02	g x^2
14	61	x
15	15 01	g FRAC
16	23 01	STO 1
17	24 00	RCL 0
18	61	x
19	14 01	f INT
20	23 02	STO 2
21	24 03	RCL 3
22	33	EEX
23	02	2
24	71	÷

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	51	+
26	23 04	STO 4
27	74	R/S
28	24 02	RCL 2
29	24 03	RCL 3
30	51	+
31	13 43	GTO 43
32	24 02	RCL 2
33	24 03	RCL 3
34	41	−
35	13 43	GTO 43
36	24 02	RCL 2
37	24 03	RCL 3
38	61	x
39	13 43	GTO 43
40	24 02	RCL 2
41	24 03	RCL 3
42	71	÷
43	14 71	f x=y
44	13 01	GTO 01
45	24 04	RCL 4
46	32	CHS
47	13 27	GTO 27
48		
49		

REGISTRES	
R_0	Max.
R_1	Nombres au hasard
R_2	Numéro de gauche
R_3	Numéro de droite
R_4	Problème
R_5	
R_6	
R_7	

Exemple:

Soit: Max = 12 et s = 0.725

Solution:

[f] [PRGM] [R/S] ——→ 6.01
(6 + 1 = 7)

7 [R/S] ——→ 8.03
(8 x 3 = 25)

25 [GTO] [3] [6] [R/S] ——→ –8.03
Essayer de nouveau: 8 × 3 = 24

24 [GTO] [3] [6] [R/S] ——→ 3.11
(3 – 11 = –8)

8 [CHS] [GTO] [3] [2] [R/S] ——→ 9.00
(9 + 0 = 9)

9 [R/S] ——→ 2.05
etc.

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Stocker Max ($0 \leq$ Max ≤ 100)		STO	0			
3	Stocker	s	STO	1			
4	Choisir le format d'affichage		f	FIX	2		
5	Commencer le problème		f	PRGM	R/S		$n_1 . n_2$
6	Choisir une opération et frap-						
	per votre réponse au clavier						
	Pour une addition (+)	$n_1 + n_2$	R/S				
	Pour une soustraction (−)	$n_1 - n_2$	GTO	32	R/S		
	Pour une multiplication (×)	$n_1 \times n_2$	GTO	36	R/S		
	Pour une division (÷)	$n_1 \div n_2$	GTO	40	R/S		
7	Si votre réponse est juste,						
	le programme affiche un nou-						
	veau problème. Aller en 6						$n_3 . n_4$
8	Si votre réponse est fausse, le						
	programme affiche le même						
	problème. Aller en 6						$-n_1 . n_2$
9	Effectuer les opérations 6 à 8						
	autant de fois que vous le souhaitez						
10	Pour changer la valeur de Max, aller en 2, puis en 5						

CHAPITRE 4: NAVIGATION

NAVIGATION ORTHODROMIQUE ET LOXODROMIQUE

Les longs voyages en mer ou dans l'air suivent toujours deux types de routes: l'orthodromie et la loxodromie. La trajectoire reliant deux points du globe en coupant tous les méridiens à angle constant est la loxodromie. Sur une projection de Mercator, la trajectoire est une droite passant par les deux points donnés. Du fait du cap constant, cette route est très pratique et souvent utilisée pour les traversées courtes à des latitudes moyennes ou faibles.

Pour d'autres latitudes, la traversée la plus courte suit l'arc de grand cercle, c'est-à-dire la trajectoire orthodromique. Malgré tout, cette route idéale est impossible à suivre dans la pratique, puisque le cap change continuellement. L'arc de grand cercle idéal sera donc interpolé par une succession de segments loxodromiques.

Le premier programme calcule des points intermédiaires sur l'arc de grand cercle théorique, connaissant les latitudes et longitudes du point de départ et du point de destination de la traversée, puis une série de longitudes quelconques intermédiaire λ_i. Pour chaque λ_i, le programme calcule la latitude L_i du point correspondant sur l'arc de grand cercle.

Plusieurs points intermédiaires (L_i, λ_i) étant obtenus, le second programme calcule la route loxodromique entre chaque point. Les données de ce programme sont les coordonnées de deux points du globe. Les résultats sont la distance et l'angle de la loxodromie. Le second programme peut être utilisé seul ou en liaison avec le programme précédent.



POINTS INTERMÉDIAIRES SUR L'ARC DE GRAND CERCLE

Formules:

$$L_i = \text{arc tg}\left[\frac{\text{tg } L_2 \sin(\lambda_i - \lambda_1) - \text{tg } L_1 \sin(\lambda_i - \lambda_2)}{\sin(\lambda_2 - \lambda_1)}\right]$$

où (L_1, λ_1): coordonnées du point de départ
(L_2, λ_2): coordonnées du point de destination
(L_i, λ_i): coordonnées d'un point intermédiaire

Remarque:

Le programme ne fonctionne pas pour des longitudes $\lambda_1 = \lambda_2$.

AFFICHAGE PAS	CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00			λ_i, D.MS				
01	15 00	g →H	λ_i, D.d				(Conversion de λ_i en
02	23 04	STO 4	λ_i				degrés décimaux)
03	24 01	RCL 1	λ_1	λ_i			
04	41	–	$\lambda_i - \lambda_1$				
05	14 04	f SIN	sin_1				$sin_1 = \sin(\lambda_i - \lambda_1)$
06	24 02	RCL 2	L_2	sin_1			
07	14 06	f TAN	tan_2	sin_1			$tg_2 = tg\ L_2$
08	61	x	$tan_2\ sin_1$				
09	24 04	RCL 4	λ_i	$tan_2\ sin_1$			
10	24 03	RCL 3	λ_2	λ_i	$tan_2\ sin_1$		
11	41	–	$\lambda_i - \lambda_2$	$tan_2\ sin_1$			
12	14 04	f SIN	sin_2	$tan_2\ sin_1$			$sin_2 = \sin(\lambda_i - \lambda_2)$
13	24 00	RCL 0	L_1	sin_2	$tan_2\ sin_1$		
14	14 06	f TAN	tan_1				$tg_1 = tg\ L_1$
15	61	x	$tan_1\ sin_2$	$tan_2\ sin_1$			
16	41	–	NUM				NUM $= tg_2\ sin_1 - tg_1\ sin_2$
17	24 03	RCL 3	λ_2	NUM			
18	24 01	RCL 1	λ_1	λ_2	NUM		
19	41	–	$\lambda_2 - \lambda_1$	NUM			
20	14 04	f SIN	DEN	NUM			DEN $= \sin(\lambda_2 - \lambda_1)$
21	71	÷	NUM/DEN				
22	15 06	g TAN^{-1}	L_i, D.d				
23	14 00	f →H.MS	L_i, D.MS				Affichage de L_i en D.MS
24	14 11 04	f FIX 4					
25	13 00	GTO 00					
26							
27							
28							
29							
30							
31							
32							
33							
34							
35							
36							
37							
38							
39							
40							
41							
42							
43							
44							
45							
46							
47							
48							
49							

REGISTRES	
R_0	L_1 (deg. déc.)
R_1	λ_1 (deg. déc.)
R_2	L_2 (deg. déc.)
R_3	λ_2 (deg. déc.)
R_4	λ_i (deg. déc.)
R_5	
R_6	
R_7	

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire les coordonnées						
	du point de départ:						
	latitude (CHS pour Sud)	L_1, D.MS	g	→H	STO	0	L_1, deg. déc.
	longitude (CHS pour Est)	λ_1, D.MS	g	→H	STO	1	λ_1, deg. déc.
3	Introduire les coordonnées du point de destination:						
	latitude (CHS pour Sud)	L_2, D.MS	g	→H	STO	2	L_2, deg. déc.
	longitude (CHS pour Est)	λ_2, D.MS	g	→H	STO	3	λ_2, deg. déc.
4	Revenir en début de mémoire		f	PRGM			
5	Introduire la longitude intermé-						
	diaire (CHS pour Sud) et cal-						
	culer la latitude correspondante	λ_i, D.MS	R/S				L_i, D.MS
6	Pour d'autre λ_i, aller en 5.						
	Pour d'autres points de départ						
	(ou de destination), aller en 2						
	(ou en 3).						

NAVIGATION LOXODROMIQUE

Formules:

$$C = \text{arc tg} \frac{\pi(\lambda_1 - \lambda_2)}{180\,[\ln \text{tg}\,(45 + \frac{1}{2}L_2) - \ln \text{tg}\,(45 + \frac{1}{2}L_1)]}$$

$$D = \begin{cases} 60\,(\lambda_2 - \lambda_1)\cos L\,;\ \cos C = 0 \\ 60\,\dfrac{(L_2 - L_1)}{\cos C}\,;\ \text{autrement} \end{cases}$$

où (L_1, λ_1): coordonnées du point de départ
(L_2, λ_2): coordonnées du point de destination
C: angle loxodromique
D: distance loxodromique

Remarques:

1. Le programme n'accepte pas les routes passant par les pôles.
2. La route ne doit pas traverser le méridien 180° (limite des heures internationales).
3. Lorsque C est très proche de 90° ou de 270°, les distances peuvent être incorrectes.
4. La précision est moins bonne pour des courses très courtes.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00			λ_2	λ_1			
01	41	−	$\lambda_1 - \lambda_2$				
02	23 06	STO 6	$\lambda_1 - \lambda_2$				
03	02	2	2	$\lambda_1 - \lambda_2$			
04	71	÷	α				$\alpha = 1/2\,(\lambda_1 - \lambda_2)$
05	14 04	f SIN	sin α				Ramène α tel que
06	15 04	g SIN^{-1}	α				$-180 \leqslant \lambda_1 - \lambda_2 \leqslant 180$
07	09	9	9	α			Détermine la route la plus
08	00	0	90	α			courte
09	71	÷	$\alpha/90$				
10	15 73	g π	π	$\alpha/90$			
11	61	×	$\pi\alpha/90$	$\pi\alpha/90$			
12	24 05	RCL 5	ln tan_2	$\pi\alpha/90$			
13	24 04	RCL 4	ln tan_1	y			$y = \pi\,\alpha/90$
14	41	−	x	y			$x = \ln tg_2 - \ln tg_1$
15	15 09	g →P	r	C			C = arc tg y/x
16	22	R↓	C			r	
17	15 03	g ABS	$\lvert C\rvert$			r	
18	23 07	STO 7	$\lvert C\rvert$			r	
19	24 06	RCL 6	$\lambda_1 - \lambda_2$	$\lvert C\rvert$			
20	14 04	f SIN	sin 2α	$\lvert C\rvert$			Ramène $\lambda_1 - \lambda_2$ tel que
21	15 04	g SIN^{-1}	2α	$\lvert C\rvert$			$-90 \leqslant \lambda_1 - \lambda_2 \leqslant 90$
22	15 41	g x<0	2α	$\lvert C\rvert$			x < 0 signifie Est → Ouest
23	13 26	GTO 26	2α	$\lvert C\rvert$			
24	21	x⇄y	$\lvert C\rvert$	2α			Ouest → Est, $\lvert C\rvert$ est la réponse
25	13 31	GTO 31	$\lvert C\rvert$	2α			
26	03	3	3	2α	$\lvert C\rvert$		Est → Ouest, la réponse est
27	06	6	36	2α	$\lvert C\rvert$		$360 - \lvert C\rvert$
28	00	0	360	2α	$\lvert C\rvert$		
29	24 07	RCL 7	$\lvert C\rvert$	360	2α	$\lvert C\rvert$	
30	41	−	$360 - \lvert C\rvert$				
31	74	R/S					Affichage de l'angle
32	06	6	6				Calcul de la distance D
33	00	0	60				
34	24 07	RCL 7	$\lvert C\rvert$	60			
35	14 05	f COS	cos $\lvert C\rvert$	60			
36	15 61	g x≠0	cos $\lvert C\rvert$	60			Si cos C≠0,
37	13 45	GTO 45	cos $\lvert C\rvert$	60			aller au pas 45
38	34	CLX	0	60			cos C = 0; le cap est
39	24 06	RCL 6	$\lambda_1 - \lambda_2$				Ouest ou Est
40	61	×	$60\,(\lambda_1 - \lambda_2)$				
41	24 02	RCL 2	L_2	$60\,(\lambda_1 - \lambda_2)$			
42	14 05	f COS	cos L_2	$60\,(\lambda_1 - \lambda_2)$			
43	61	×	Distance				$D = 60\,(\lambda_1 - \lambda_2)\cos L$
44	13 00	GTO 00	Distance				Affichage de la distance
45	71	÷	$60/\cos \lvert C\rvert$				Le cap n'est pas Est ou Ouest
46	24 02	RCL 2	L_2				Appliquer la formule
47	24 00	RCL 0	L_1	L_2	$60/\cos \lvert C\rvert$		$D = 60\,(L_2 - L_1)/\cos C$
48	41	−	$L_2 - L_1$	$60/\cos \lvert C\rvert$			
49	61	×	Distance				Stop

REGISTRES	
R_0	L_1 (deg. déc.)
R_1	λ_1 (deg. déc.)
R_2	L_2 (deg. déc.)
R_3	λ_2 (deg. déc.)
R_4	ln tg $(45+L_1/2)$
R_5	ln tg $(45+L_2/2)$
R_6	$\lambda_1 - \lambda_2$
R_7	$\lvert C\rvert$

Exemple:

Un navigateur désire se rendre de San Francisco (L: 37°49′N, λ: 122°25′O) à Tokyo (L: 35°40′N, λ: 139°45′E) en suivant trois routes loxodromiques pour interpoler un arc de grand cercle. Les longitudes des deux points intermédiaires sont 155°O et 175°E. Calculer pour chaque étape l'angle de route et la distance.

Solution:

Calcul de la latitude de 2 points intermédiaires:

37.49 g →H STO 0 122.25 g →H STO 1 35.40 g →H STO 2 139.45 CHS g →H STO 3 f PRGM 155 R/S ⟶ 47.4606

175 CHS R/S ⟶ 47.3610

Coordonnées de 2 points intermédiaires: (L: 47°46′N, λ: 155°O) et (L: 47°36′N, λ: 175°E).

Calcul des loxodromies:

Coordonnées du point de départ:

37.49 g →H STO 2 2 ÷ 45 + f tan f ln STO 5
122.25 g →H STO 3

Calcul de l'angle et de la distance au premier point intermédiaire:

47.4606 g →H RCL 2 STO 0 x⇄y STO 2 2 ÷ 45 + f tan f ln RCL 5 STO 4 x⇄y STO 5 155 g →H RCL 3 STO 1 x⇄y STO 3 f PRGM

R/S ⟶ 292.67 (angle)

R/S ⟶ 1549.38 (distance)

Calcul de l'angle et de la distance au deuxième point intermédiaire:

47.361 g →H RCL 2 STO 0 x⇄y STO 2 2 ÷ 45 + f tan f ln RCL 5 STO 4 x⇄y STO 5 175 CHS g →H RCL 3 STO 1 x⇄y STO 3 f PRGM

R/S ⟶ 269.53 (angle)

R/S ⟶ 1211.80 (distance)

Calcul de l'angle et de la distance au point de destination:

35.40 g →H RCL 2 STO 0 x⇄y STO 2 2 ÷ 45 + f tan f ln RCL 5 STO 4 x⇄y STO 5 139.45 CHS g →H RCL 3 STO 1 x⇄y STO 3 f PRGM

R/S ⟶ 245.53 (angle)

R/S ⟶ 1728.66 (distance)

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire la latitude initiale						
	(CHS pour Sud)	L_1, D.MS	g	→H	STO	2	
			2	÷	45	+	
			f	TAN	f	LN	
			STO	5			ln tg_1
3	Introduire la longitude initiale						
	(CHS pour Est)	λ_1, D.MS	g	→H	STO	3	λ_1, deg. déc.
4	Introduire la latitude finale						
	(CHS pour Sud)	L_2, D.MS	g	→H	RCL	2	
			STO	0	x⇄y	STO	
			2	2	÷	45	
			+	f	TAN	f	
			LN	RCL	5	STO	
			4	x⇄y	STO	5	ln tg_2
5	Introduire la longitude finale						
	(CHS pour Est)	λ_2, D.MS	g	→H	RCL	3	
			STO	1	x⇄y	STO	
			3				λ_2, deg. déc.
6	Calcul de l'angle		f	PRGM	R/S		C
7	Calcul de la distance		R/S				D
8	Pour enchaîner sur un autre						
	point, retourner en 4.						

En résumé:

Position	Coordonnées	Loxodromie Angle	Loxodromie Distance
San Francisco	L: 37°49′N, λ: 122°25′O		
		292.7°	1549.16 m.n.
Premier point intermédiaire	L: 47°46′N, λ: 155°O		
		269.5°	1211.81 m.n.
Deuxième point intermédiaire	L: 47°36′N, λ: 175°E		
		245.5°	1728.51 m.n.
Tokyo	L: 35°40′N, λ: 139°45′E		



La distance totale des 3 routes loxodromiques est égale à 4489.5 miles nautiques, tandis que celle suivant un arc de grand cercle est égale à 4460 miles nautiques. Pour deux points intermédiaire, la distance supplémentaire à parcourir en navigation loxodromique est inférieure à 30 miles nautiques.

RÉSOLUTION DU TRIANGLE DE POSITION

Ce programme donne la hauteur calculée Hc et l'azimuth Zn d'un astre, la latitude L de l'observateur, l'angle horaire local LHA et la déclinaison de l'astre étant connus. Il remplace les 9 tables HO 214.

Formules: $Hc = \arcsin [\sin d \sin \mathbf{L} + \cos d \cos \mathbf{L} \cos \mathbf{LHA}]$

$$Z = \arccos \left[\frac{\sin d - \sin L \sin Hc}{\cos L \cos Hc} \right]$$

$$Zn = \begin{cases} Z & ; \sin LHA < 0 \\ 360\text{-}Z; & \sin LHA \geqslant 0 \end{cases}$$

Remarques:

1. Introduire les latitudes Sud et les déclinaisons Sud comme des valeurs négatives.
2. Vous pouvez introduire l'angle méridien t à la place de l'angle horaire local LHA; dans ce cas, introduire les angles méridiens Est comme des valeurs négatives.

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	24 00	RCL 0
02	14 04	f SIN
03	24 01	RCL 1
04	14 04	f SIN
05	61	x
06	24 00	RCL 0
07	14 05	f COS
08	24 01	RCL 1
09	14 05	f COS
10	61	x
11	24 02	RCL 2
12	14 05	f COS
13	61	x
14	51	+
15	23 03	STO 3
16	15 04	g SIN^{-1}
17	23 04	STO 4
18	14 00	f →H.MS
19	74	R/S
20	24 01	RCL 1
21	14 04	f SIN
22	24 03	RCL 3
23	24 00	RCL 0
24	14 04	f SIN

AFFICHAGE		TOUCHES
PAS	CODE	
25	61	x
26	41	–
27	24 00	RCL 0
28	14 05	f COS
29	71	÷
30	24 04	RCL 4
31	14 05	f COS
32	71	÷
33	15 05	g COS^{-1}
34	24 02	RCL 2
35	14 04	f SIN
36	15 41	g x<0
37	13 45	GTO 45
38	22	R↓
39	03	3
40	06	6
41	00	0
42	21	x⇄y
43	41	–
44	13 00	GTO 00
45	22	R↓
46	13 00	GTO 00
47		
48		
49		

REGISTRES
R_0 L
R_1 d
R_2 LHA
R_3 sin Hc
R_4 Hc
R_5
R_6
R_7

Exemple:

Calculer la hauteur et l'azimuth de la lune.
Angle horaire local LHA: 2°39′54″O
Déclinaison d: 13°51′06″S
Latitude L: 33°20′N

Solution:

Hc = 42°44′4″
Zn = 183.5°

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire:						
	la latitude de l'observateur	L, D.MS	g	→H	STO	0	L, deg. déc.
	la déclinaison	d, D.MS	g	→H	STO	1	d, deg. déc.
	l'angle horaire local	LHA, D.MS	g	→H	STO	2	LHA, deg. déc.
3	Calcul de la hauteur		f	PRGM	R/S		Hc, D.MS
4	Calcul de l'azimuth		R/S				Zn, deg. déc.
5	Pour un nouveau cas, aller en 2						

NAVIGATION SUIVANT UN ARC DE GRAND CERCLE

Ce programme calcule l'arc de grand cercle entre deux points et le cap initial à suivre, la latitude et la longitude du point de départ (L_1, λ_1) et du point de destination (L_2, λ_2) étant connues.

Formules:

$$D = 60 \text{ arc cos } [\sin L_1 \sin L_2 + \cos L_1 \cos L_2 \cos(\lambda_2 - \lambda_1)]$$

$$H = \text{arc cos} \left[\frac{\sin L_2 - \sin L_1 \cos(D/60)}{\sin(D/60) \cos L_1} \right]$$

$$H_i = \begin{cases} H & ; \sin(\lambda_2 - \lambda_1) < 0 \\ 360\text{-}H ; & \sin(\lambda_2 - \lambda_1) \geqslant 0 \end{cases}$$

Remarques:

1. Introduire les latitudes Sud et les longitudes Est comme des valeurs négatives.
2. Erreurs d'arrondi lorsque le point de départ et le point de destination sont très proches l'un de l'autre (1 mile au moins).
3. Ne pas introduire des coordonnées de points du globe diamétralement opposés.
4. Ne pas introduire les latitudes de +90° ou −90°.
5. Ne pas calculer un cap initial lorsque les longitudes sont égales ($L_1 = L_2$).

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	24 00	RCL 0
02	14 04	f SIN
03	24 01	RCL 1
04	14 04	f SIN
05	61	x
06	24 00	RCL 0
07	14 05	f COS
08	24 01	RCL 1
09	14 05	f COS
10	61	x
11	24 02	RCL 2
12	14 05	f COS
13	61	x
14	51	+
15	23 03	STO 3
16	15 05	g COS^{-1}
17	23 04	STO 4
18	06	6
19	00	0
20	61	x
21	74	R/S
22	24 01	RCL 1
23	14 04	f SIN
24	24 00	RCL 0

AFFICHAGE		TOUCHES
PAS	CODE	
25	14 04	f SIN
26	24 03	RCL 3
27	61	x
28	41	–
29	24 00	RCL 0
30	14 05	f COS
31	71	÷
32	24 04	RCL 4
33	14 04	f SIN
34	71	÷
35	15 05	g COS^{-1}
36	24 02	RCL 2
37	14 04	f SIN
38	15 41	g $x<0$
39	13 47	GTO 47
40	22	R↓
41	03	3
42	06	6
43	00	0
44	21	$x \rightleftarrows y$
45	41	–
46	13 00	GTO 00
47	22	R↓
48	13 00	GTO 00
49		

REGISTRES
R_0 L_1
R_1 L_2
R_2 $\lambda_2 - \lambda_1$
R_3 cos (D/60)
R_4 D/60
R_5
R_6
R_7

Exemple:

Un navigateur désire se rendre de San Francisco (L: 37°49′N, λ: 22°25′O) à Tokyo (L: 35°40′N, λ: 139°45′E). Calculer la distance suivant un arc de grand cercle entre ces deux villes et le cap initial à suivre.

Solution:

D = 4460.04
H_i = 303.29°

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire:						
	latitude du point de départ	L_1, D.MS	g	→H	STO	0	L_1, deg. déc.
	latitude du point de destination	L_2, D.MS	g	→H	STO	1	L_2, deg. déc.
	longitude du point de destination	λ_2, D.MS	g	→H			λ_2, deg. déc.
	longitude du point de départ	λ_1, D.MS	g	→H	–	STO	
			2				$\lambda_2 - \lambda_1$, deg.déc.
3	Calcul de la distance suivant un arc de grand cercle		f	PRGM	R/S		D, milles naut.
4	Calcul du cap initial		R/S				H_i, deg. déc.
5	Pour un nouveau cas, aller en 2.						

CHAPITRE 5: CALCULS NUMÉRIQUES

SOLUTION DE L'ÉQUATION f(x)=0 PAR LA MÉTHODE DE NEWTON

Pour résoudre une équation telle que $\ln x + 3x = 10.8074$, il n'existe pas de solution algébrique simple. Dans de nombreux cas, plusieurs algorithmes permettent de résoudre l'équation $f(x)=0$, où $f(x)=\ln x + 3x - 10.8074$.

Ce programme utilise la méthode de Newton pour résoudre l'équation $f(x)=0$, dans laquelle $f(x)$ est donnée. 14 pas de programme sont prévus pour mémoriser la séquence de touches nécessaire au calcul de $f(x)$, x étant dans le registre X. Les registres de la pile opérationnelle et les registres mémoire R_5 à R_8 sont aussi disponibles. D'autre part, il faut définir une estimation initiale x_1.

Le programme s'arrête quand deux approximations successives x_i et x_{i+1} sont calculées avec une tolérance ε , c'est-à-dire quand $|x_{i+1}-x_i|<\varepsilon$. Vous devez introduire la valeur de ε ($10^{-6}x_1$ est une valeur raisonnable de ε).

Formules:

La méthode de Newton utilise la formule suivante pour calculer les approximations suivantes:

$$x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)}$$



Ce programme utilise la dérive f′(x) selon la formule suivante:

$$x_{i+1} = x_i - \delta_i \left[\frac{f(x_i + \delta_i)}{f(x_i)} - 1 \right]^{-1}$$

où $\delta_i = 10^{-5} x_i$

Remarques:

1. A la fin du calcul, la dernière valeur de f(x) peut être affichée au moyen des touches RCL 4. Si vous désirez obtenir une autre valeur de f(x) plus près de 0, exécutez à nouveau le programme avec une valeur de ε plus petite.
2. Vous pouvez vérifier la convergence vers zéro de la fonction en modifiant légèrement le programme. Pour cela, remplacer l'instruction g NOP au pas 07 par une instruction f PAUSE : le programme s'arrêtera pendant 1 seconde environ durant chaque itération, affichant les valeurs convergentes vers zéro de f(x). Pour effectuer cette modification dans un programme déjà enregistré:

   1. Appuyer sur GTO 0 6
   2. Passer en mode PRGM
   3. Appuyer sur f PAUSE
   4. Passer en mode RUN
   5. Appuyer sur f PRGM

Remarques sur la programmation:

Ce programme est un des plus complexes de ce fascicule. A chaque itération, les fonctions f(x) et f(x + δ) doivent toutes les deux être calculées, mais la fonction f est introduite en mémoire seulement une fois. Les calculateurs disposant d'une plus grande capacité de mémoire résolvent ce problème au moyen d'un sous-programme. Ce programme utilise une astuce (9 pas de programme) en mettant en mémoire dans le registre R_0 une variable qui simule le résultat d'un test logique.
Après le calcul de f, le test suivant est effectué:
Variable à l'état 0: branchement du programme à une instruction qui mettra en mémoire f(x).
Variable à l'état 1: calcul d'une dérivée basée sur f(x + δ).

AFFICHAGE PAS	CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00							
01	34	CLX	0				Mise du flag à 0 pour f(x)
02	23 00	STO 0	0				
03	24 01	RCL 1	x	0			Rappel de x et branchement
04	13 17	GTO 17	x	0			pour calcul de f(x)
05	22	R↓	f(x)				Permutation pour enlever
06	23 04	STO 4	f(x)				le flag
07	15 74	g NOP	f(x)				
08	01	1	1	f(x)			
09	23 00	STO 0	1	f(x)			Mise du flag à 1 pour f(x + δ)
10	24 01	RCL 1	x	1	f(x)		
11	24 01	RCL 1	x	x	1	f(x)	
12	33	EEX	1. 00	x	x	1	
13	05	5	1. 05	x	x	1	
14	71	÷	10^{-5} x	x	1	1	
15	23 03	STO 3	δ	x	1	1	
16	51	+	x + δ	1	1	1	
17							
18							
19							Définition de f(x)
20							par l'utilisateur
21							Partie du programme réservée
22							au calcul de f(x) et de
23							f(x + δ). Flag dans R_0:
24							état 0 pour f(x), état 1 pour
25							f(x + δ)
26							
27							
28							
29							
30							
31	15 71	g x = 0	f(x)/(x + δ)				La valeur de la fonction est-
32	13 49	GTO 49	f(x)/(x + δ)				elle égale à 0?
33	24 00	RCL 0	Flag	f(x)/(x + δ)			Oui, solution
34	15 71	g x = 0	Flag	f(x)/(x + δ)			Non, voir flag
35	13 05	GTO 05	Flag	f(x)			Flag = 0?
36	22	R↓	f(x + δ)			Flag	Oui, calcul de f(x)
37	24 04	RCL 4	f(x)	f(x + δ)			Non, flag = 1, calcul de f(x + δ)
38	71	÷	R				R = f(x + δ)/f(x)
39	01	1	1	R			
40	41	−	R − 1				R − 1 = [f(x + δ) − f(x)]/f(x)
41	15 22	g 1/x	$(R-1)^{-1}$				Approximation de
42	24 03	RCL 3	δ	$(R-1)^{-1}$			f′(x) = [f(x + δ) − f(x)]/δ
43	61	×	δ/(R − 1)				Δ = f(x)/f′(x)
44	23 41 01	STO − 1	Δ				$x_i+1 = x_i - \Delta$
45	15 03	g ABS	|Δ|				x_i+1
46	24 02	RCL 2	ε	|Δ|			x_i+1
47	14 41	f x<y	ε	|Δ|			$\|x_i+1 - x_i\| > \varepsilon$?
48	13 01	GTO 01	ε	|Δ|			Oui, nouvelle itération
49	24 01	RCL 1	x	ε	|Δ|		Non, affichage de x et arrêt.

REGISTRES	
R_0	Flag
R_1	x
R_2	ε
R_3	δ
R_4	f(x)
R_5	
R_6	
R_7	

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire les pas 1 à 16 du programme						16 51
2	Introduire la fonction f(x)						
3	Effectuer un branchement au pas 31		GTO	31			
4	Appuyer sur SST jusqu'à						
	obtenir l'affichage du pas 30						
5	Introduire les pas 31 à 49 du programme						
6	Passer en mode RUN						
7	Mettre en mémoire l'estimation initiale	x_1	STO	1			
8	Mettre en mémoire la tolérance	ϵ	STO	2			
9	Calcul de la solution		f	PRGM	R/S		x_0
10	Pour modifier x_1 ou ε, aller au pas correspondant,						
	puis mettre en mémoire la nouvelle valeur.						

Exemple:

Les constructeurs d'engrenage ont fréquemment à résoudre l'équation involute: $\mathrm{tg}x-x-I=0$ dans laquelle x est un angle exprimé en radians et I l'involute de x. Quel est l'angle x_0 correspondant à une involute de 0.0324?

Remarque:

Si vous souhaitez modifier souvent la valeur de I, mettez cette valeur en mémoire dans le registre R_7.

Solution:

$x_0 = 25.62°$
$f(x_0) = 0.00$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire les pas 1 à 16 du programme						16 51
2	Introduire						
	$f(x)=\mathrm{tg}x-x-I$		f	TAN			17 14 06
			f	LASTx			18 14 73
			−				19 41
			RCL	7			20 24 07
			−				21 41
3	Effectuer le branchement au pas 31		GTO	31			22 13 31
4	Appuyer 8 fois sur SST:						
	affichage du pas 30						
5	Introduire les pas 31 à 49 du programme						49 24 01
6	Passer en mode RUN						
7	Passer en mode angulaire		g	RAD			
8	Mettre en mémoire I	.0324	STO	7			
9	Mettre en mémoire l'estimation $x_1=1$	1	STO	1			
10	Mettre en mémoire la tolérance ($\varepsilon=10^{-6}$)	10^{-6}	STO	2			
11	Calcul de x_0		f	PRGM	R/S		0.45
12	Convertir l'angle en degrés		180	x	g	π	
			÷				25.62
13	Affichage de la dernière valeur de f(x)		RCL	4			0.00

INTÉGRATION NUMÉRIQUE PAR LA MÉTHODE DE SIMPSON

Soient $x_0, x_1, \ldots, x_n$ des points également répartis tels que $x_i = x_0 + ih$ pour $i = 0, 1, 2, \ldots, n$ et $f(x_0), f(x_1), \ldots, f(x_n)$ les valeurs correspondantes de f(x). Il n'est pas nécessaire que la fonction soit connue explicitement; si elle l'est, il suffit de la programmer dans le HP-25 pour obtenir les divers points de la fonction. n doit être un entier positif pair.

La méthode de Simpson est la suivante:

$$\int_{x_0}^{x_n} f(x)\,dx \cong \frac{h}{3}\,[f(x_0) + 4f(x_1) + 2f(x_2) + \ldots + 4f(x_{n-3}) + 2f(x_{n-2}) + 4f(x_{n-1}) + f(x_n)]\,.$$

La réponse est indiquée par I.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	24 00	RCL 0
02	03	3
03	71	÷
04	23 00	STO 0
05	61	x
06	23 01	STO 1
07	74	R/S
08	24 00	RCL 0
09	61	x
10	24 01	RCL 1
11	51	+
12	23 01	STO 1
13	74	R/S
14	24 00	RCL 0
15	61	x
16	04	4
17	61	x
18	24 01	RCL 1
19	51	+
20	23 01	STO 1
21	74	R/S
22	24 00	RCL 0
23	61	x
24	02	2
25	61	x
26	24 01	RCL 1
27	51	+
28	23 01	STO 1
29	13 13	GTO 13
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 h/3
R_1 Σ
R_2
R_3
R_4
R_5
R_6
R_7

Exemple:

Calculer $\int_0^{\pi} \sin^2 x\,dx$ au moyen de la méthode de Simpson avec $h = \pi/8$

Déterminer au préalable les données suivantes:

i	0	1	2	3	4	5	6	7	8
x_i	0	$\pi/8$	$\pi/4$	$3\pi/8$	$\pi/2$	$5\pi/8$	$3\pi/4$	$7\pi/8$	π
$f(x_i)$	0	0.1464	0.5	0.8536	1	0.8536	0.5	0.1464	0

Solution:

$$\int_0^{\pi} \sin^2 x\,dx \simeq 1.5708$$

La solution exacte est $\pi/2$.

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire l'incrément	h	STO	0			
3	Introduire $f(x_0)$	$f(x_0)$	f	PRGM	R/S		Somme partielle
4	Introduire $f(x_n)$	$f(x_n)$	R/S				Somme partielle
5	Introduire les valeurs $i = 1, 2 \ldots n-2$	$f(x_i)$	R/S				Somme partielle
6	Introduire la valeur $i = n-1$	$f(x_{n-1})$	R/S				I

ÉQUATION DIFFÉRENTIELLE DU PREMIER ORDRE

Ce programme peut être utilisé pour résoudre les équations différentielles de la forme:

$y' = f(x, y)$

avec des valeurs initiales x_0, y_0.

La méthode employée est numérique et calcule y_i pour $x_i = x_0 + ih$ ($i = 1, 2, 3, \ldots$), h étant l'incrément fixé par l'utilisateur.

Ce programme utilise la méthode d'Euler:

$$\hat{y}_{i+1} = y_i + h f(x_i, y_i)$$

$$y_{i+1} = y_i + \frac{h}{2}\left[f(x_i, y_i) + f(x_{i+1}, \hat{y}_{i+1})\right]$$

f(x, y) est introduit en mémoire à partir du pas 18. Vous disposez de 13 pas de programme pour écrire la fonction f(x, y) et des registres mémoires R_5, R_6 et R_7. x et y se trouvent respectivement dans les registres X et Y de la pile opérationnelle. Ce programme doit afficher la valeur de f(x, y) dans le registre X et se terminer par GTO 31.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	34	CLX
02	23 04	STO 4
03	24 02	RCL 2
04	24 01	RCL 1
05	13 18	GTO 18
06	22	R↓
07	23 03	STO 3
08	24 00	RCL 0
09	61	x
10	24 02	RCL 2
11	51	+
12	24 01	RCL 1
13	24 00	RCL 0
14	51	+
15	01	1
16	23 04	STO 4
17	22	R↓
18		
19		
20		
21		
22		
23		
24		
25		
26		
27		
28		
29		
30		
31	24 04	RCL 4
32	15 71	g x=0
33	13 06	GTO 06
34	22	R↓
35	24 03	RCL 3
36	51	+
37	24 00	RCL 0
38	61	x
39	02	2
40	71	÷
41	24 02	RCL 2
42	51	+
43	23 02	STO 2
44	24 01	RCL 1
45	24 00	RCL 0
46	51	+
47	23 01	STO 1
48	14 74	f PAUSE
49	22	x⇄y

REGISTRES
R_0 h
R_1 x
R_2 y
R_3 y′
R_4 Flag
R_5
R_6
R_7

Exemple:

Calculer l'équation différentielle suivante: $y' = x\sqrt{y}$ avec comme conditions initiales $x_0 = 1$ et $y_0 = 1$. $h = 0.1$

Solution:

Pour introduire la fonction, appuyer sur les touches x⇄y f √x ×.

x	1.0	1.1	1.2	1.3	1.4	1.5
y (programme)	1.0	1.1077	1.2319	1.3745	1.5372	1.7221
y (exact)	1.0	1.1078	1.2321	1.3748	1.5376	1.7227

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire les pas 1 à 17 du programme						17 22
2	Introduire la fonction f(x, y)						
3	Effectuer le branchement au pas 31		GTO	31			
4	Appuyer sur SST jusqu'à						
	l'affichage du pas 30						
5	Introduire les pas 31 à 49 du programme						49 13 01
6	Passer en mode RUN						
7	Mettre en mémoire l'incrément	h	STO	0			
8	Introduire les valeurs	x_0	STO	1			
	initiales	y_0	STO	2	f	PRGM	
9	Affichage de la valeur						
	suivante de x et de la		R/S				(x_k)
	valeur y correspondante						y_k
10	Répéter 9 pour d'autres valeurs						

INTERPOLATION LINÉAIRE

Si $[(x_1, f(x_1)]$ et $[(x_2, f(x_2)]$ sont deux points d'une fonction $f(x)$, la fonction en x_0 peut être approximée par la formule suivante:

$$f(x_0) \cong \frac{(x_2-x_0)\, f(x_1) + (x_0-x_1)\, f(x_2)}{(x_2-x_1)}$$

Cette formule est celle de l'interpolation linéaire. x_2 est toujours différent de x_1.

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	23 04	STO 4
02	24 00	RCL 0
03	41	−
04	24 03	RCL 3
05	61	x
06	24 02	RCL 2
07	24 04	RCL 4
08	41	−
09	24 01	RCL 1
10	61	x
11	51	+
12	24 02	RCL 2
13	24 00	RCL 0
14	41	−
15	71	÷
16	13 00	GTO 00
17		
18		
19		
20		
21		
22		
23		
24		

AFFICHAGE		TOUCHES
PAS	CODE	
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 x_1
R_1 $f(x_1)$
R_2 x_2
R_3 $f(x_2)$
R_4 x_0
R_5
R_6
R_7

Exemple:

Soit: $f(7.3) = 1.9879$
$f(7.4) = 2.0015$,
Interpolation linéaire $f(7.37)$.

Solution:

$f(7.37) = 1.9974$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire	x_1	STO	0			
	le premier point	$f(x_1)$	STO	1			
3	Mettre en mémoire	x_2	STO	2			
	le deuxième point	$f(x_2)$	STO	3	f	PRGM	
4	Introduire x_0; calcul de $f(x_0)$	x_0	R/S				$f(x_0)$
5	Répéter 5 pour d'autres						
	valeurs de x						

CHAPITRE 6: STATISTIQUES

AJUSTEMENT DE COURBE – RÉGRESSION LINÉAIRE



Lors de la recherche d'une formule expérimentale pour interpréter un phénomène, il faut d'abord effectuer une série d'observations de deux caractères (x, y). A première vue, la relation entre x et y paraît linéaire, c'est-à-dire que l'équation est de la forme $y = ax + b$ avec a et b constants. Ce programme calcule par la méthode des moindres carrés les constantes a et b qui lient au mieux les données expérimentales à l'équation $y = a_1x + a_0$.

Introduire d'abord les valeurs de tous les couples de données (x_i, y_i), $i = 1 \ldots n$. Le HP-25 calculera alors les constantes de régression a_1 et a_0 et éventuellement le coefficient de détermination r^2. Le coefficient r^2 mesure le degrés de perfection de l'ajustement de la droite de régression. La valeur de ce coefficient est comprise entre 0 et 1; si $r^2 = 1$, l'ajustement est idéal.

Equations:

$y = a_1 x + a_0$

Toutes les sommations ci-dessous sont effectuées pour $i = 1, \ldots n$.

$$a_1 = \frac{\Sigma xy - \dfrac{\Sigma x \, \Sigma y}{n}}{\Sigma x^2 - \dfrac{(\Sigma x)^2}{n}}$$

$$a_0 = \bar{y} - a_1 \bar{x}$$

où
$$\bar{y} = \frac{\Sigma y}{n}$$

$$\bar{x} = \frac{\Sigma x}{n}$$

Coefficient de détermination:

$$r^2 = \frac{\left[\Sigma xy - \dfrac{\Sigma x \Sigma y}{n}\right]^2}{\left[\Sigma x^2 - \dfrac{(\Sigma x)^2}{n}\right]\left[\Sigma y^2 - \dfrac{(\Sigma y)^2}{n}\right]}$$

Remarque:

Les valeurs de a_0 et a_1 sont respectivement contenues dans les registres mémoires R_0 et R_1. Après calcul de a_0, a_1 et r^2, la valeur estimée de y, $\hat{y}$, correspondante à n'importe quelle valeur de x peut être calculée au moyen de l'équation $y = a_1 x + a_0$.

Remarque sur la programmation:

La valeur intermédiaire $C = \Sigma xy - (\Sigma x \times \Sigma y / n)$ est d'abord calculée au pas 14; néanmoins, cette valeur est nécessaire en fin de programme pour le calcul de r^2. Tous les registres mémoire R_0 à R_7 étant utilisés, la valeur de C est conservée dans la pile opérationnelle jusqu'au pas 36; ne pas modifier les contenus de la pile après le calcul de a_0 et a_1 (voir mode opératoire – instruction N° 4).

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00			y	x			Pas 1 à 7 pour sommation
01	31	↑	y	y	x		
02	15 02	g x^2	y^2	y	x		
03	23 51 02	STO + 2	y^2	y	x		Σy^2
04	22	R↓	y	x		y^2	
05	21	x⇄y	x	y		y^2	
06	25	Σ+	n	y		y^2	n, Σy, Σxy, Σx^2, Σx
07	13 00	GTO 00	n	y		y^2	
08	24 05	RCL 5	Σxy				
09	24 07	RCL 7	Σx	Σxy			
10	24 04	RCL 4	Σy	Σx	Σxy		
11	61	×	Σx Σy	Σxy			
12	24 03	RCL 3	n	Σx Σy	Σxy		
13	71	÷	Σx Σy/n	Σxy			
14	41	–	C				C = Σxy – (Σx Σy/n)
15	24 06	RCL 6	Σx^2	C			
16	24 07	RCL 7	Σx	Σx^2	C		
17	15 02	g x^2	$(\Sigma x)^2$	Σx^2	C		
18	24 03	RCL 3	n	$(\Sigma x)^2$	Σx^2	C	
19	71	÷	$(\Sigma x)^2/n$	Σx^2	C	C	
20	41	–	D	C	C	C	$D = \Sigma x^2 - [(\Sigma x)^2/n]$
21	71	÷	a_1	C	C	C	$a_1 = C/D$
22	23 01	STO 1	a_1	C	C	C	
23	24 07	RCL 7	Σx	a_1	C	C	
24	61	×	a_1 Σx	C	C	C	
25	32	CHS	$-a_1$ Σx	C	C	C	
26	24 04	RCL 4	Σy	$-a_1$ Σx	C	C	
27	51	+	Σy – a_1 Σx	C	C	C	
28	24 03	RCL 3	n	Σy – a_1 Σx	C	C	
29	71	÷	a_0	C	C	C	$a_0 = \bar{y} - a_1 \bar{x}$
30	23 00	STO 0	a_0	C	C	C	
31	74	R/S	a_0	C	C	C	Affichage de a_0
32	24 01	RCL 1	a_1	a_0	C	C	
33	74	R/S	a_1	a_0	C	C	Affichage de a_1
34	21	x⇄y	a_0	a_1	C	C	
35	22	R↓	a_1	C	C	a_0	
36	61	×	a_1 C	C	a_0	a_0	
37	24 02	RCL 2	Σy^2	a_1 C	C	a_0	
38	24 04	RCL 4	Σy	Σy^2	a_1 C	C	
39	15 02	g x^2	$(\Sigma y)^2$	Σy^2	a_1 C	C	
40	24 03	RCL 3	n	$(\Sigma y)^2$	Σy^2	a_1 C	
41	71	÷	$(\Sigma y)^2/n$	Σy^2	a_1 C	a_1 C	
42	41	–	E	a_1 C	a_1 C	a_1 C	$E = \Sigma y^2 - [(\Sigma y)^2/n]$
43	71	÷	r^2	a_1 C	a_1 C	a_1 C	$r^2 = a_1 C/E$
44	13 00	GTO 00	r^2	a_1 C	a_1 C	a_1 C	
45							
46							
47							
48							
49							

REGISTRES	
R_0	a_0
R_1	a_1
R_2	Σy^2
R_3	n
R_4	Σy
R_5	Σxy
R_6	Σx^2
R_7	Σx

Exemple:

Lors d'un contrôle de qualité, un ingénieur constate une relation entre le volume d'un produit chimique ajouté à un lot de la concentration finale de ce produit dans le produit final ($y = a_1x + a_0$). Les données suivantes représentent le poids en gramme ajouté (x) et le poids dans le produit final (y):

x	3	1	5	5	7	8	8.5
y	2	1	6	3	7	6	9

Calculer les valeurs de a_1 et a_0 ainsi que le coefficient de détermination.

Solution:

[f] [PRGM] [f] [REG] [3] [1] [2] [R/S] → 1.00
[1] [↑] [1] [R/S] → 2.00
[5] [↑] [6] [R/S] → 3.00
[5] [↑] [3] [R/S] → 4.00
[7] [↑] [7] [R/S] → 5.00
[8] [↑] [6] [R/S] → 6.00
[8.5] [↑] [9] [R/S] → 7.00
[GTO] [0] [8] [R/S] → 1.22
[R/S] → 0.85
[R/S] → 0.83

D'où l'équation $y = 0.85x + 1.22$. Le coefficient de détermination r^2 est égal à 0.83.

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	REG	f	PRGM	
3	Introduire les valeurs de x et						
	de y pour i = 1, ... n	x_i	↑				
		y_i	R/S				i
4	Calcul des constantes		GTO	08	R/S		a_0*
	de régression		R/S				a_1*
5	Calcul du coefficient						
	de détermination		R/S				r^2
6	Introduire x;						
	estimation de $\hat{y}$	x	RCL	1	x	RCL	
			0	+			$\hat{y}$
7	Pour une autre estimation de $\hat{y}$,						
	aller en 6						
8	Pour un nouveau cas,						
	aller en 2						
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

AJUSTEMENT D'UNE FONCTION EXPONENTIELLE

Ce programme calcule l'ajustement d'un nombre n de paires de points $\{(x_i, y_i), i=1, 2 \ldots, n\}$ par la méthode des moindres carrés, avec $y_i>0$ à l'aide d'une fonction exponentielle du type:

$y=a\,e^{bx}$ $(a>0)$.

Cette équation se linéarise par:

$\ln y=\ln a+bx$.

Le programme calcule les éléments suivants:

1. Coefficients a, b:

$$b=\frac{\Sigma x_i \ln y_i-\frac{1}{n}(\Sigma x_i)(\Sigma \ln y_i)}{\Sigma x_i^2-\frac{1}{n}(\Sigma x_i)^2}$$

$$a=\exp\left[\frac{\Sigma \ln y_i}{n}-b\frac{\Sigma x_i}{n}\right]$$

2. Coefficient de détermination

$$r^2=\frac{\left[\Sigma x_i \ln y_i-\frac{1}{n}\Sigma x_i\,\Sigma \ln y_i\right]^2}{\left[\Sigma x_i^2-\frac{(\Sigma x_i)^2}{n}\right]\left[\Sigma(\ln y_i)^2-\frac{(\Sigma \ln y_i)^2}{n}\right]}$$

3. La valeur estimée $\hat{y}$ pour x donné: $\hat{y}=a\,e^{bx}$

Remarque:

n est un entier positif différent de 1.

AFFICHAGE PAS	CODE	TOUCHES
00		
01	14 07	f LN
02	31	↑
03	15 02	g x^2
04	23 51 02	STO + 2
05	22	R↓
06	21	x⇄y
07	25	Σ+
08	13 00	GTO 00
09	24 05	RCL 5
10	24 07	RCL 7
11	24 04	RCL 4
12	61	x
13	24 03	RCL 3
14	71	÷
15	41	–
16	24 06	RCL 6
17	24 07	RCL 7
18	15 02	g x^2
19	24 03	RCL 3
20	71	÷
21	41	–
22	71	÷
23	23 01	STO 1
24	24 07	RCL 7

AFFICHAGE PAS	CODE	TOUCHES
25	61	x
26	32	CHS
27	24 04	RCL 4
28	51	+
29	24 03	RCL 3
30	71	÷
31	15 07	g e^x
32	23 00	STO 0
33	74	R/S
34	24 01	RCL 1
35	74	R/S
36	21	x⇄y
37	22	R↓
38	61	x
39	24 02	RCL 2
40	24 04	RCL 4
41	15 02	g x^2
42	24 03	RCL 3
43	71	÷
44	41	–
45	71	÷
46	13 00	GTO 00
47		
48		
49		

REGISTRES
R_0 a
R_1 b
R_2 $\Sigma(\ln y)^2$
R_3 n
R_4 $\Sigma \ln y$
R_5 $\Sigma x \ln y$
R_6 Σx^2
R_7 Σx

Exemple:

x_i	0.72	1.31	1.95	2.58	3.14
y_i	2.16	1.61	1.16	0.85	0.5

Solution:

$a=3.45$, $b=-0.58$

$y=3.45\,e^{-0.58x}$

$r^2=0.98$

Pour $x=1.5$, $\hat{y}=1.44$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1							
2			f	REG	f	PRGM	
3							
		x_i	↑				
		y_i	R/S				i
4	Calcul des constantes		GTO	09	R/S		a*
			R/S				b*
5	Calculs du coefficient						
	de détermination		R/S				r^2
6	Introduire x,	x	RCL	1	x	g	
	estimation de $\hat{y}$		e^x	RCL	0	x	$\hat{y}$
7	Pour une autre estimation de $\hat{y}$						
	aller en 6						
8	Pour un nouveau cas,						
	aller en 2						
	*Après ce résultat, ne pas modifier le contenu de la pile.						

AJUSTEMENT D'UNE FONCTION LOGARITHMIQUE

Ce programme ajuste une fonction logarithmique

$y = a + b \ln x$

à un ensemble de points

$\{(x_i, y_i), i = 1, 2, ..., n\}$

avec $x_i > 0$.

Il calcule:

1. Les coefficients de régression

$$b = \frac{\Sigma y_i \ln x_i - \frac{1}{n} \Sigma \ln x_i \Sigma y_i}{\Sigma (\ln x_i)^2 - \frac{1}{n} (\Sigma \ln x_i)^2}$$

$$a = \frac{1}{n} (\Sigma y_i - b \Sigma \ln x_i)$$

2. Le coefficient de détermination

$$r^2 = \frac{\left[\Sigma y_i \ln x_i - \frac{1}{n} \Sigma \ln x_i \Sigma y_i\right]^2}{\left[\Sigma (\ln x_i)^2 - \frac{1}{n} (\Sigma \ln x_i)^2\right] \left[\Sigma y_i^2 - \frac{1}{n} (\Sigma y_i)^2\right]}$$

3. La valeur estimée $\hat{y}$ pour x donné $\hat{y} = a + b \ln x$

Remarque:

n est un entier positif différent de 1.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	31	$\uparrow$
02	15 02	g x^2
03	23 51 02	STO + 2
04	22	$R\downarrow$
05	21	$x \rightleftarrows y$
06	14 07	f LN
07	25	$\Sigma+$
08	13 00	GTO 00
09	24 05	RCL 5
10	24 07	RCL 7
11	24 04	RCL 4
12	61	x
13	24 03	RCL 3
14	71	÷
15	41	–
16	24 06	RCL 6
17	24 07	RCL 7
18	15 02	g x^2
19	24 03	RCL 3
20	71	÷
21	41	–
22	71	÷
23	23 01	STO 1
24	24 07	RCL 7

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	61	x
26	32	CHS
27	24 04	RCL 4
28	51	+
29	24 03	RCL 3
30	71	÷
31	23 00	STO 0
32	74	R/S
33	24 01	RCL 1
34	74	R/S
35	21	$x \rightleftarrows y$
36	22	$R\downarrow$
37	61	x
38	24 02	RCL 2
39	24 04	RCL 4
40	15 02	g x^2
41	24 03	RCL 3
42	71	÷
43	41	–
44	71	÷
45	13 00	GTO 00
46		
47		
48		
49		

REGISTRES
R_0 a
R_1 b
R_2 Σy^2
R_3 n
R_4 Σy
R_5 $\Sigma y \ln x$
R_6 $\Sigma \ln x$
R_7 $\Sigma (\ln x)^2$

Exemple:

x_i	3	4	6	10	12
y_i	1.5	9.3	23.4	45.8	60.1

Solution:

$a = -47.02$, $b = 41.39$
$y = -47.02 + 41.39 \ln x$
$r^2 = 0.98$
Pour $x = 8$, $\hat{y} = 39.06$
Pour $x = 14.5$, $\hat{y} = 63.67$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	REG	f	PRGM	
3	Introduire les valeurs de x						
	et de y pour i = 1, ... n	x_i	$\uparrow$				
		y_i	R/S				i
4	Calcul des constantes		GTO	09	R/S		a*
			R/S				b*
5	Calcul du coefficient						
	de détermination		R/S				r^2
6	Introduire x;	x	f	ln	RCL	1	
	estimation de $\hat{y}$		x	RCL	0	+	$\hat{y}$
7	Pour une autre estimation						
	de $\hat{y}$, aller en 6						
8	Pour un nouveau cas,						
	aller en 2.						
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

AJUSTEMENT D'UNE FONCTION PUISSANCE

Ce programme ajuste une fonction puissance

$$y = a\,x^b \ (a>0)$$

à un ensemble de points

$$(x_i, y_i),\ i = 1, 2, ..., n\}$$

avec $x_i>0$, $y_i>0$

Si on linéarise cette équation de la manière suivante

$$\ln y = b \ln x + \ln a$$

le problème peut être résolu comme un problème d'ajustement linéaire.

Eléments calculés par le programme:

1. Coefficients de régression

$$b = \frac{\Sigma (\ln x_i)(\ln y_i) - \dfrac{(\Sigma \ln x_i)(\Sigma \ln y_i)}{n}}{\Sigma (\ln x_i)^2 - \dfrac{(\Sigma \ln x_i)^2}{n}}$$

$$a = \exp\left[\frac{\Sigma \ln y_i}{n} - b\,\frac{\Sigma \ln x_i}{n}\right]$$

2. Coefficient de détermination

$$r^2 = \frac{\left[\Sigma (\ln x_i)(\ln y_i) - \dfrac{(\Sigma \ln x_i)(\Sigma \ln y_i)}{n}\right]^2}{\left[\Sigma (\ln x_i)^2 - \dfrac{(\Sigma \ln x_i)^2}{n}\right]\left[\Sigma (\ln y_i)^2 - \dfrac{(\Sigma \ln y_i)^2}{n}\right]}$$

3. La valeur estimée $\hat{y}$ pour x donné: $\hat{y} = ax^b$

Remarque:

n est un entier positif différent de 1.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	14 07	f LN
02	31	↑
03	15 02	g x^2
04	23 51 02	STO + 2
05	22	R↓
06	21	x⇄y
07	14 07	f LN
08	25	Σ+
09	13 00	GTO 00
10	24 05	RCL 5
11	24 07	RCL 7
12	24 04	RCL 4
13	61	x
14	24 03	RCL 3
15	71	÷
16	41	–
17	24 06	RCL 6
18	24 07	RCL 7
19	15 02	g x^2
20	24 03	RCL 3
21	71	÷
22	41	–
23	71	÷
24	23 01	STO 1

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	24 07	RCL 7
26	61	x
27	32	CHS
28	24 04	RCL 4
29	51	+
30	24 03	RCL 3
31	71	÷
32	15 07	g e^x
33	23 00	STO 0
34	74	R/S
35	24 01	RCL 1
36	74	R/S
37	21	x⇄y
38	22	R↓
39	61	x
40	24 02	RCL 2
41	24 04	RCL 4
42	15 02	g x^2
43	24 03	RCL 3
44	71	÷
45	41	–
46	71	÷
47	13 00	GTO 00
48		
49		

REGISTRES
R_0 a
R_1 b
R_2 $\Sigma (\ln y)^2$
R_3 n
R_4 $\Sigma \ln y$
R_5 $\Sigma (\ln x)(\ln y)$
R_6 $\Sigma (\ln x)^2$
R_7 $\Sigma \ln x$

Exemple:

x_i	10	12	15	17	20	22	25	27	30	32	35
y_i	0.95	1.05	1.25	1.41	1.73	2.00	2.53	2.98	3.85	4.59	6.02

Solution:

$a = .03$, $b = 1.46$
$y = .03x^{1.46}$
$r^2 = .94$
Pour $x = 18$, $\hat{y} = 1.76$
Pour $x = 23$, $\hat{y} = 2.52$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	REG	f	PRGM	
3	Introduire les valeurs de x						
	et de y pour i=1, ... n	x_i	↑				
		y_i	R/S				i
4	Calcul des constantes		GTO	10	R/S		a*
			R/S				b*
5	Calcul du coefficient						
	de détermination		R/S				r^2
6	Introduire x;	x	RCL	1	f	y^x	
	estimation de $\hat{y}$		RCL	0	x		$\hat{y}$
7	Pour une autre estimation						
	de $\hat{y}$, aller en 6						
8	Pour un nouveau cas,						
	aller en 2						
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

COVARIANCE ET COEFFICIENT DE CORRÉLATION

Soit une suite de valeurs données $\{(x_i, y_i), i = 1, 2, ..., n\}$, la covariance et le coefficient de corrélation sont définis par:

covariance $$s_{xy} = \frac{1}{n-1}\left(\Sigma x_i y_i - \frac{1}{n}\Sigma x_i \Sigma y_i\right)$$

ou $$s_{xy}' = \frac{1}{n}\left(\Sigma x_i y_i - \frac{1}{n}\Sigma x_i \Sigma y_i\right)$$

coefficient de corrélation $$r = \frac{s_{xy}}{s_x s_y}$$

s_x et s_y étant l'écart type

$$s_x = \sqrt{\frac{\Sigma x_i^2 - (\Sigma x_i)^2/n}{n-1}} \qquad s_y = \sqrt{\frac{\Sigma y_i^2 - (\Sigma y_i)^2/n}{n-1}}$$

Remarque:

$-1 \leqslant r \leqslant 1$

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	31	↑
02	15 02	g x^2
03	23 51 02	STO + 2
04	22	R↓
05	21	x⇄y
06	25	Σ+
07	13 00	GTO 00
08	24 05	RCL 5
09	24 04	RCL 4
10	24 07	RCL 7
11	61	x
12	24 03	RCL 3
13	71	÷
14	41	–
15	24 03	RCL 3
16	01	1
17	41	–
18	23 00	STO 0
19	71	÷
20	23 01	STO 1
21	74	R/S
22	24 00	RCL 0
23	61	x
24	24 03	RCL 3

AFFICHAGE		TOUCHES
PAS	CODE	
25	71	÷
26	74	R/S
27	14 22	f s
28	23 71 01	STO ÷ 1
29	24 02	RCL 2
30	24 04	RCL 4
31	15 02	g x^2
32	24 03	RCL 3
33	71	÷
34	41	–
35	24 00	RCL 0
36	71	÷
37	14 02	f $\sqrt{x}$
38	23 71 01	STO ÷ 1
39	24 01	RCL 1
40	13 00	GTO 00
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 n − 1
R_1 Utilisé
R_2 Σy^2
R_3 n
R_4 Σy
R_5 Σxy
R_6 Σx^2
R_7 Σx

Exemple:

x_i	26	30	44	50	62	68	74
y_i	92	85	78	81	54	51	40

Solution:

$S_{xy} = -354.14$
$S_{xy}' = -303.55$
$r = -0.96$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM	f	REG	
3	Effectuer 3 pour i=1, 2, ... n	x_i	↑				
		y_i	R/S				i
4	Calcul de la covariance s_{xy}		GTO	08	R/S		s_{xy}
5	Calcul de s_{xy}'		R/S				s_{xy}'
6	Calcul du coefficient de corrélation		R/S				r
7	Pour un nouveau cas, aller en 2.						

MOMENTS ET COEFFICIENTS D'ASYMÉTRIE

Ce programme effectue les calculs statistiques suivants pour une suite de valeur $\{x_1, x_2, \ldots, x_n\}$:

Moment d'ordre 1 $$\overline{x} = \frac{1}{n}\sum_{i=1}^{n} x_i$$

Moment d'ordre 2 $$m_2 = \frac{1}{n}\ \Sigma x_i^2 - \overline{x}^2$$

Moment d'ordre 3 $$m_3 = \frac{1}{n}\ \Sigma x_i^3 - \frac{3}{n}\ \overline{x}\ \Sigma x_i^2 + 2\overline{x}^3$$

Coefficient d'asymétrie $$\gamma_1 = \frac{m_3}{m_2^{3/2}}$$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	31	↑
02	15 02	g x^2
03	25	Σ+
04	13 00	GTO 00
05	24 04	RCL 4
06	24 03	RCL 3
07	71	÷
08	23 02	STO 2
09	74	R/S
10	24 07	RCL 7
11	24 03	RCL 3
12	71	÷
13	24 02	RCL 2
14	15 02	g x^2
15	41	−
16	23 01	STO 1
17	74	R/S
18	24 05	RCL 5
19	24 03	RCL 3
20	71	÷
21	24 07	RCL 7
22	24 02	RCL 2
23	61	x
24	24 03	RCL 3

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	71	÷
26	03	3
27	61	x
28	41	−
29	24 02	RCL 2
30	31	↑
31	15 02	g x^2
32	61	x
33	02	2
34	61	x
35	51	+
36	23 00	STO 0
37	74	R/S
38	24 00	RCL 0
39	24 01	RCL 1
40	01	1
41	73	•
42	05	5
43	14 03	f y^x
44	71	÷
45	13 00	GTO 00
46		
47		
48		
49		

REGISTRES
R_0 m_3
R_1 m_2
R_2 $\overline{x}$
R_3 n
R_4 Σx
R_5 Σx^3
R_6 Σx^4
R_7 Σx^2

Exemple:

i	1	2	3	4	5	6	7	8	9
x_i	2.1	3.5	4.2	6.5	4.1	3.6	5.3	3.7	4.9

Solution:

$\bar{x} = 4.21$
$m_2 = 1.39$
$m_3 = 0.39$
$\gamma_1 = 0.24$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM	f	REG	
3	Effectuer 3 pour i=1, 2 ... n						
		x_i	R/S				i
4	Effacer la donnée incorrecte	x_k	↑	g	x^2	f	
			Σ-				
5	Calcul de la moyenne		GTO	05	R/S		$\bar{x}$
6	Calcul du moment m_2 d'ordre 2						
	et du moment m_3 d'ordre 3		R/S				m_2
			R/S				m_3
7	Calcul du coefficient						
	d'asymétrie		R/S				γ_1
8	Pour un nouveau cas, aller en 2						

DISTRIBUTION NORMALE

Une distribution normale type est représentée par la fonction:

$$f(x) = \frac{1}{\sqrt{2\pi}} e^{-\frac{x^2}{2}} .$$

la surface de droite étant

$$Q(x) = \frac{1}{\sqrt{2\pi}} \int_x^{\infty} e^{-\frac{t^2}{2}} dt .$$



Pour $x \geqslant 0$, le programme calcule Q(x) par la formule d'approximation polynomiale:

$Q(x) = f(x)\,(b_1 t + b_2 t^2 + b_3 t^3 + b_4 t^4 + b_5 t^5) + \varepsilon(x)$

avec $|\varepsilon(x)| < 7.5 \times 10^{-8}$

$t = \frac{1}{1 + rx}$, $r = 0.2316419$

$b_1 = 0.31938153$, $b_2 = -0.356563782$
$b_3 = 1.781477937$, $b_4 = -1.821255978$
$b_5 = 1.330274429$

Remarque:

Dans ce programme, x doit être $\geqslant 0$. Les équations $f(-x) = f(x)$, $Q(-x) = 1 - Q(x)$ avec $x \geqslant 0$, peuvent être utilisées pour f et Q pour les nombres négatifs.

Référence:

Handbook of Mathematical Functions, Abramowitz and Stegun, National Bureau of Standards, 1968.

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	31	↑
02	23 06	STO 6
03	61	x
04	02	2
05	71	÷
06	32	CHS
07	15 07	g e^x
08	15 73	g π
09	02	2
10	61	x
11	14 02	f $\sqrt{x}$
12	71	÷
13	23 07	STO 7
14	74	R/S
15	24 00	RCL 0
16	24 06	RCL 6
17	61	x
18	01	1
19	51	+
20	15 22	g 1/x
21	31	↑
22	31	↑
23	31	↑
24	24 05	RCL 5

AFFICHAGE		TOUCHES
PAS	CODE	
25	61	x
26	24 04	RCL 4
27	51	+
28	61	x
29	24 03	RCL 3
30	51	+
31	61	x
32	24 02	RCL 2
33	51	+
34	61	x
35	24 01	RCL 1
36	51	+
37	61	x
38	24 07	RCL 7
39	61	x
40	13 00	GTO 00
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 r
R_1 b_1
R_2 b_2
R_3 b_3
R_4 b_4
R_5 b_5
R_6 x
R_7 f(x)

Exemples:

1. x = 1.18
2. x = 2.28

Solutions:

1. f(x) = 0.20
   Q(x) = 0.12
2. f(x) = 0.03
   Q(x) = 0.01

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM			
3	Mettre en mémoire les	r	STO	0			
	constantes	b_1	STO	1			
		b_2	STO	2			
		b_3	STO	3			
		b_4	STO	4			
		b_5	STO	5			
4	Introduire x; calcul de f(x)	x	R/S				f(x)
5	Calcul de Q(x)		R/S				Q(x)
6	Pour un nouveau cas, aller en 4.						

BORNE INFÉRIEURE DE L'INTÉGRALE D'UNE DISTRIBUTION NORMALE

Ce programme détermine la valeur de x telle que:

$$Q = \int_x^{\infty} \frac{e^{-\frac{t^2}{2}}}{\sqrt{2\pi}}\, dt$$

avec Q donné tel que $0<Q\leqslant 0.5$.



On utilise la formule d'approximation suivante:

$$x = t - \frac{c_0 + c_1\, t + c_2\, t^2}{1 + d_1\, t + d_2\, t^2 + d_3\, t^3} + \epsilon(Q)$$

Avec $|\epsilon(Q)| < 4.5 \times 10^{-4}$

$$t = \sqrt{\ln \frac{1}{Q^2}}$$

$c_0 = 2.515517$ $\quad d_1 = 1.432788$
$c_1 = 0.802853$ $\quad d_2 = 0.189269$
$c_2 = 0.010328$ $\quad d_3 = 0.001308$

Référence:

Handbook of Mathematical Functions, Abramowitz and Stegun, National Bureau of Standards, 1968.

AFFICHAGE PAS	CODE	TOUCHES	AFFICHAGE PAS	CODE	TOUCHES
00			25	51	+
01	31	↑	26	61	x
02	61	x	27	24 00	RCL 0
03	15 22	g 1/x	28	51	+
04	14 07	f LN	29	24 07	RCL 7
05	14 02	f $\sqrt{x}$	30	71	÷
06	23 06	STO 6	31	41	−
07	31	↑	32	13 00	GTO 00
08	31	↑	33		
09	31	↑	34		
10	24 05	RCL 5	35		
11	61	x	36		
12	24 04	RCL 4	37		
13	51	+	38		
14	61	x	39		
15	24 03	RCL 3	40		
16	51	+	41		
17	61	x	42		
18	01	1	43		
19	51	+	44		
20	23 07	STO 7	45		
21	34	CLX	46		
22	24 02	RCL 2	47		
23	61	x	48		
24	24 01	RCL 1	49		

REGISTRES
R_0 c_0
R_1 c_1
R_2 c_2
R_3 d_1
R_4 d_2
R_5 d_3
R_6 t
R_7 $1 + d_1 t + d_2 t^2 + d_3 t^3$

Exemples:

1. Q=0.12
2. Q=0.05

Solutions:

1. x=1.18
2. x=1.65

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM			
3	Mettre en mémoire	c_0	STO	0			
	les constantes	c_1	STO	1			
		c_2	STO	2			
		d_1	STO	3			
		d_2	STO	4			
		d_3	STO	5			
4	Introduire Q	Q	R/S				x
5	Pour un nouveau cas, aller en 4						

FACTORIELLE

Ce programme calcule les factorielles de nombres entiers positifs compris entre 2 et 69. Le programme de la fonction Gamma pourrait également calculer la factorielle, mais nécessiterait davantage de pas de programme.

$n! = n(n-1)\,(n-2)\,\ldots\,(2)\,(1)$

Remarques:

1. Plus les valeurs de n sont grandes, plus le calculateur met de temps pour donner le résultat (environ 20 secondes pour $n = 69$).
2. Le programme ne vérifie pas les valeurs introduites; le calculateur donnera des résultats incorrects pour des valeurs de n non entières, <2 et >69.

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	31	↑
02	01	1
03	23 00	STO 0
04	21	x⇄y
05	23 61 00	STO×0
06	01	1
07	41	–
08	14 61	f x≠y
09	13 05	GTO 05
10	24 00	RCL 0
11	13 00	GTO 00
12		
13		
14		
15		
16		
17		
18		
19		
20		
21		
22		
23		
24		

AFFICHAGE		TOUCHES
PAS	CODE	
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 Utilisé
R_1
R_2
R_3
R_4
R_5
R_6
R_7

Exemples:

1. $5! = 120.00$
2. $10! = 3628800.00$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM			
3	Introduire n($2 \leq n \leq 69$)	n	R/S				n!
4	Pour un nouveau cas, aller en 3.						

ARRANGEMENT

Un arrangement est un sous-ensemble ordonné d'un ensemble d'objets distincts. Le nombre d'arrangements possibles, chacun contenant n objets, qui peuvent être réalisés à partir d'un ensemble de m objets distincts, est donné par:

$$ {}_mA_n = \frac{m!}{(m-n)!} = m(m-1)\ldots(m-n+1) $$

où m et n sont des entiers tels que $0 \leqslant n \leqslant m$

Remarques:

1. ${}_mA_n$ peut être désigné par A_n^m, A (m, n) ou $(m)_n$.
2. ${}_mA_0 = 1$, ${}_mA_1 = m$, ${}_mA_m = m!$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	24 00	RCL 0
02	24 00	RCL 0
03	24 01	RCL 1
04	15 71	g x=0
05	13 29	GTO 29
06	14 71	f x=y
07	13 31	GTO 31
08	14 51	f x⩾y
09	13 39	GTO 39
10	01	1
11	14 71	f x=y
12	13 41	GTO 41
13	22	R↓
14	41	−
15	01	1
16	51	+
17	61	x
18	14 73	f LASTx
19	24 00	RCL 0
20	01	1
21	41	−
22	14 71	f x=y
23	13 26	GTO 26
24	22	R↓

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	13 15	GTO 15
26	22	R↓
27	22	R↓
28	13 00	GTO 00
29	01	1
30	13 00	GTO 00
31	01	1
32	41	−
33	15 71	g x=0
34	13 37	GTO 37
35	23 61 00	STO x 0
36	13 31	GTO 31
37	24 00	RCL 0
38	13 00	GTO 00
39	00	0
40	71	÷
41	22	R↓
42	22	R↓
43	13 00	GTO 00
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 m
R_1 n
R_2
R_3
R_4
R_5
R_6
R_7

Exemples:

1. ${}_{43}A_3 = 74046.00$
2. ${}_{73}A_4 = 26122320.00$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire m et n	m	STO	0			
		n	STO	1			
3	Calcul de l'arrangement		f	PRGM	R/S		${}_mA_n$
4	Pour un nouveau cas, aller en 2.						

COMBINAISON

Une combinaison est une sélection non ordonnée d'un ou plusieurs ensembles d'objets distincts. Le nombre de combinaisons possibles, chacune contenant n objets, est donné par:

$$_mC_n = \frac{m!}{(m-n)!\,n!} = \frac{m(m-1)\ldots(m-n+1)}{1\cdot 2\cdot \ldots \cdot n}$$

où m et n sont des entiers tels que $0 \leqslant n \leqslant m$

Ce programme calcule $_mC_n$ en utilisant l'algorithme suivant:

1. Si $n \leqslant m-n$

$$_mC_n = \frac{m-n+1}{1}\cdot\frac{m-n+2}{2}\cdot\ldots\cdot\frac{m}{n}.$$

2. Si $n > m-n$, le programme calcule $_mC_{m-n}$.

Remarques:

1. $_mC_n$, qui est aussi appelé coefficient binomial, peut être désigné par C_n^m, $C(m, n)$, ou $\binom{m}{n}$.
2. $_mC_n = {}_mC_{m-n}$
3. $_mC_0 = {}_mC_m = 1$
4. $_mC_1 = {}_mC_{m-1} = m$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	41	−
02	14 73	f LASTx
03	14 41	f x<y
04	21	x⇄y
05	23 00	STO 0
06	01	1
07	23 01	STO 1
08	51	+
09	23 02	STO 2
10	22	R↓
11	15 71	g x=0
12	13 30	GTO 30
13	01	1
14	24 01	RCL 1
15	51	+
16	23 01	STO 1
17	21	x⇄y
18	14 51	f x≥y
19	13 22	GTO 22
20	24 02	RCL 2
21	13 00	GTO 00
22	22	x⇄y
23	24 00	RCL 0
24	51	+

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	24 01	RCL 1
26	71	÷
27	23 61 02	STO x 2
28	22	R↓
29	13 13	GTO 13
30	01	1
31	13 00	GTO 00
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 max (n, m − n)
R_1 Utilisé
R_2 Utilisé
R_3
R_4
R_5
R_6
R_7

Exemples:

1. $_{73}C_4 = 1088430.00$
2. $_{43}C_3 = 12341.00$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Introduire m, n	m	↑				
		n	f	PRGM	R/S		$_mC_n$
3	Pour un nouveau cas, aller en 2						

GÉNÉRATEUR DE NOMBRES ALÉATOIRES

Ce programme calcule des nombres aléatoires u_i uniformément distribués tels que:

$0 \leqslant u_i \leqslant 1$

à l'aide de la formule suivante:

u_i = partie fractionnaire de $[(\pi + u_{i-1})^5]$.

L'utilisateur devra choisir le nombre initial u_0 tel que:

$0 \leqslant u_0 \leqslant 1$.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	15 73	g π
02	24 00	RCL 0
03	51	+
04	05	5
05	14 03	f y^x
06	15 01	g FRAC
07	23 00	STO 0
08	13 00	GTO 00
09		
10		
11		
12		
13		
14		
15		
16		
17		
18		
19		
20		
21		
22		
23		
24		

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 u_i
R_1
R_2
R_3
R_4
R_5
R_6
R_7

Exemple:

Calculer les nombres aléatoires uniformément distribués à partir de 0.192743568.

Solution:

0.14, 0.76, 0.15, 0.35, 0.62, 0.54, 0.62, 0.91, 0.48, 0.24, ...

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire u_0	u_0	STO	0	f	PRGM	
3	Calcul de u_i		R/S				u_i
4	Pour une autre valeur de u_i,						
	aller en 3						
5	Pour un nouveau cas, aller en 2.						

CALCUL DE LA VALEUR DU CHI-CARRÉ

Le test de l'accord global entre une «distribution observée» et une «distribution théorique» spécifiée «a priori» ou ajustée aux observations est obtenu en calculant la quantité

$$\chi^2 = \sum_{i=1}^{n} \frac{(O_i - E_i)^2}{E_i}$$

où les O_i sont les fréquences observées et les E_i les fréquences prévues pour la distribution ajustée.

Remarques:

1. Afin d'effectuer ce test sur un ensemble de données connues, il peut être nécessaire de réunir certaines classes pour être sûr que chaque valeur de la fréquence prévue ne soit pas trop petite (pas plus petite que 5).
2. Si les fréquences prévues E_i sont toutes égales à une certaine valeur E, calculer d'abord E ($E = \frac{\Sigma 0_1}{n}$), puis introduire cette valeur pour la fréquence prévue E_i.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	00	0
02	23 00	STO 0
03	23 01	STO 1
04	74	R/S
05	23 02	STO 2
06	41	–
07	15 02	g x^2
08	24 02	RCL 2
09	71	÷
10	23 51 01	STO + 1
11	24 00	RCL 0
12	01	1
13	51	+
14	23 00	STO 0
15	13 04	GTO 04
16	23 02	STO 2
17	41	–
18	15 02	g x^2
19	24 02	RCL 2
20	71	÷
21	23 41 01	STO – 1
22	24 00	RCL 0
23	01	1
24	41	–
25	23 00	STO 0
26	13 04	GTO 04
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0 n
R_1 χ^2
R_2 E_i
R_3
R_4
R_5
R_6
R_7

Exemple:

O_i	8	50	47	56	5	14
E_i	9.6	46.75	51.85	54.4	8.25	9.15

Solution:

$\lambda^2 = 4.84$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM	R/S		0.00
3	Introduire les fréquences observées et prévues pour						
	i=1, ... n						
		O_i	↑				
		E_i	R/S				i
4	Effacer les données incorrectes	O_k	↑				
		E_k	GTO	16	R/S		
5	Affichage de x^2		RCL	1			χ^2
6	Pour un nouveau cas, aller en 2.						

TEST t SUR DES PAIRES DE VARIABLES

Soit une série d'observations prises deux par deux pour deux populations normales de moyennes inconnues μ_1 et μ_2.

x_i	x_1	x_2	...	x_n
y_i	y_1	y_2	...	y_n

Si

$$D_i = x_i - y_i$$

$$\overline{D} = \frac{1}{n} \sum_{i=1}^{n} D_i$$

$$s_D = \sqrt{\frac{\Sigma D_i^2 - \frac{1}{n}(\Sigma D_i)^2}{n-1}}$$

$$s_{\overline{D}} = \frac{s_D}{\sqrt{n}}$$

Le test statistique

$$t = \frac{\overline{D}}{s_{\overline{D}}} ,$$

qui a $(n-1)$ degrés de liberté peut être utilisé pour tester l'hypothèse nulle: $H_0 : \mu_1 = \mu_2$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	41	–
02	25	Σ+
03	13 00	GTO 00
04	14 22	f s
05	24 03	RCL 3
06	14 02	f $\sqrt{x}$
07	71	÷
08	14 21	f $\bar{x}$
09	21	x⇄y
10	71	÷
11	74	R/S
12	24 03	RCL 3
13	01	1
14	41	–
15	13 00	GTO 00
16		
17		
18		
19		
20		
21		
22		
23		
24		

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1
R_2
R_3 n
R_4 Utilisé
R_5 Utilisé
R_6 ΣD_i
R_7 ΣD_i^2

Exemple:

x_i	14	17.5	17	17.5	15.4
y_i	17	20.7	21.6	20.9	17.2

Solution:

t = –7.16
df = 1.00

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	REG	f	PRGM	
3	Introduire les paires d'observations pour i = 1, ... n						
		x_i	↑				
		y_i	R/S				i
4	Effacer les données incorrectes	x_k	↑				
		y_k	–	f	Σ–		
5	Calcul de t et df		GTO	04	R/S		t
			R/S				df
6	Pour un nouveau cas, aller en 2.						

TEST t SUR DEUX MOYENNES

Supposons que $\{x_1, x_2, ..., x_{n1}\}$ et $\{y_1, y_2, ..., y_{n2}\}$ soient deux échantillons pris au hasard de deux populations normales de moyennes inconnues μ_1 et μ_2 et de variance égale et inconnue σ^2.

Ce programme permet de tester l'hypothèse nulle

$H_0: \mu_1 - \mu_2 = D$

où D est un nombre donné.

Soit

$$\bar{x} = \frac{1}{n_1} \sum_{i=1}^{n_1} x_i$$

$$\bar{y} = \frac{1}{n_2} \sum_{i=1}^{n_2} y_i$$

$$t = \frac{\bar{x} - \bar{y} - D}{\sqrt{\frac{1}{n_1} + \frac{1}{n_2}} \sqrt{\frac{\Sigma x_i^2 - n_1 \bar{x}^2 + \Sigma y_i^2 - n_2 \bar{y}^2}{n_1 + n_2 - 2}}}$$

On peut utiliser la statistique de t dont la distribution a $n_1 + n_2 - 2$ degrés de liberté pour tester l'hypothèse nulle.

AFFICHAGE PAS	CODE	TOUCHES
00		
01	24 03	RCL 3
02	23 00	STO 0
03	24 06	RCL 6
04	23 01	STO 1
05	14 21	f $\bar{x}$
06	23 02	STO 2
07	34	CLX
08	23 03	STO 3
09	23 06	STO 6
10	23 07	STO 7
11	74	R/S
12	31	↑
13	14 21	f $\bar{x}$
14	51	+
15	24 02	RCL 2
16	21	x⇄y
17	41	−
18	24 00	RCL 0
19	15 22	g 1/x
20	24 03	RCL 3
21	15 22	g 1/x
22	51	+
23	14 02	f $\sqrt{x}$
24	71	÷

AFFICHAGE PAS	CODE	TOUCHES
25	24 01	RCL 1
26	24 02	RCL 2
27	15 02	g x^2
28	24 00	RCL 0
29	61	x
30	41	−
31	24 06	RCL 6
32	51	+
33	14 21	f $\bar{x}$
34	15 02	g x^2
35	24 03	RCL 3
36	61	x
37	41	−
38	24 00	RCL 0
39	24 03	RCL 3
40	51	+
41	02	2
42	41	−
43	71	÷
44	14 02	f $\sqrt{x}$
45	71	÷
46	13 00	GTO 00
47		
48		
49		

REGISTRES
R_0 n_1
R_1 Σx^2
R_2 $\bar{x}$
R_3 n_2
R_4 Utilisé
R_5 Utilisé
R_6 Σy^2
R_7 Σy

Exemple:

x: 79, 84, 108, 114, 120, 103, 122, 120
y: 91, 103, 90, 113, 108, 87, 100, 80, 99, 54
$n_1 = 8$
$n_2 = 10$
Si $D = 0$ (c'est-à-dire $H_0: \mu_1 = \mu_2$)

Solution:

$t = 1.73$
$\overline{x} = 106.25$
$\overline{y} = 92.50$

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	REG			
3	Introduire x pour i=1, ... n						
		x_i	Σ+				i
4	Initialiser pour y		f	PRGM	R/S		0.00
5	Introduire y pour i=1, ... n						
		y_i	Σ+				i
6	Introduire D: calcul de t	D	R/S				t
7	Calcul de $\overline{x}$ et de $\overline{y}$						
			RCL	2			$\overline{x}$
			f	$\overline{x}$			$\overline{y}$
8	Pour un nouveau cas, aller en 2.						

CHAPITRE 7: TOPOGRAPHIE

CHEMINEMENT POLYGONAL ET COMPENSATION

Ce programme calcule les coordonnées compensées des stations d'une polygonation en trois étapes successives:

1. A partir des coordonnées X_0, Y_0 de la station de départ et des gisements et distances: calcul des coordonnées brutes.
2. L'introduction des coordonnées de fermeture vraies permet de connaître les écarts de fermeture en X et Y, la longueur du cheminement (Σd).
3. Le calcul des coordonnées définitives, compensées, est obtenu par réintroduction des gisements et distances.

Les formules employées sont les suivantes:



$$X_{i+1} = X_i + d \sin G - \frac{e_x \times d}{\Sigma d}$$

$$Y_{i+1} = Y_i + d \cos G - \frac{e_y \times d}{\Sigma d}$$

G et d représentent le gisement et la distance du point i au point i+1.
e_x et e_y représentent les écarts en x et en y.

Remarque:

Ce programme convient pour un cheminement tendu aussi bien que fermé, quel que soit le nombre de côtés.
Dans le cas où vous souhaitez effectuer la compensation par une autre méthode que celle du calculateur, arrêtez l'introduction du programme après l'instruction N° 27.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00							
01	15 34	g GRD					Unité d'angle = grades
02	23 04	STO 4					Stockage des
03	23 02	STO 2					coordonnées
04	22	R↓					de la station
05	23 01	STO 1					de départ
06	23 03	STO 3					
07	34	α×					
08	23 00	STO 0					$\Sigma d = 0$
09	74	R/S	d	G^t			Affichage de Y, pose de G et d
10	23 51 00	STO+0					Calcul de Σd
11	14 09	f R	ΔY	ΔX			Calcul des Δ
12	23 51 02	STO+2	ΔY				
13	22	R↓	ΔX				
14	23 51 01	STO+1	ΔX				
15	24 01	RCL 1	X				
16	74	R/S	X				Affichage de X
17	24 02	RCL 2	Y				
18	13 09	GTO 09					
19	23 41 02	STO−2	Y_F	X_F			Y calculé − Y vrai
20	22	R↓	X_F				
21	23 41 01	STO−1	X_F				X calculé − X vrai
22	24 01	RCL 1	e_x				
23	74	R/S	e_x				Affichage de e_x
24	24 02	RCL 2	e_y				
25	74	R/S	e_y				Affichage de e_y
26	24 00	RCL 0	Σd				
27	74	R/S	Σd				Affichage de Σd,
28	23 05	STO 5	d	G			pose de G et d
29	14 09	f R	ΔY	ΔX			Affichage de Y compensé
30	24 02	RCL 2	e_y				
31	24 05	RCL 5	d	e_y	ΔY	ΔX	Calcul de Δ Y
32	61	×	de_y	ΔY	ΔX		compensé
33	24 00	RCL 0	Σd	de_y	ΔY	ΔX	
34	71	÷	C_y	ΔY	ΔX		
35	41	−	ΔY_c	ΔX			
36	23 51 04	STO+4	ΔY_c	ΔX			
37	22	R↓	ΔX				
38	24 01	RCL 1	e_x	ΔX			
39	24 05	RCL 5	d	e_x	ΔX		
40	61	×	de_x	ΔX			Calcul de Δ X
41	24 00	RCL 0	Σd	de_x	ΔX		compensé
42	71	÷	C_x	ΔX			
43	41	−	ΔX_c				
44	23 51 03	STO+3	ΔX_c				
45	24 03	RCL 3	X_c				Affichage de X compensé
46	74	R/S	X_c				
47	24 04	RCL 4	Y_c				
48	13 27	GTO 27	Y_c				
49							

REGISTRES	
R_0	Σd
R_1	X brut e_x
R_2	Y brut e_y
R_3	X compensé
R_4	Y compensé
R_5	d
R_6	
R_7	

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Coordonnées de la						
	station de départ	X_o	↑				
		Y_o	f		R/S		0.00
3	Gisement	G^t	↑				
4	Distance	d	R/S				X brut
5	Lire X brut		R/S				Y brut
6	Lire Y brut						
7	Retourner en 3						
	en fin de cheminement						
8	X fermeture	X_F	↑				
	Y fermeture	Y_F	GTO	1	9	R/S	e_x
9	Lire e_x					R/S	e_y
10	Lire e_y					R/S	Σd
11	Lire la longueur du cheminement						
12	Gisement	G^t	↑				
13	Distance	d	R/S				X compensé
14	Lire X compensé		R/S				Y compensé
15	Lire Y compensé						
16	Retourner en 12						

Exemple:



X = 123.78
Y = 751.32

X_0	123.78	↑			
Y_0	751.32	f PRGM R/S			
$G_{0\to1}$	53.714	↑			
d	75.31	R/S	180.05	x_1	brut
		R/S	801.38	y_1	brut
$G_{1\to2}$	121.312	↑			
d	81.45	R/S	256.97	x_2	brut
		R/S	774.62	y_2	brut
$G_{2\to3}$	85.731	↑			
d	62.58	R/S	317.99	x_3	brut
		R/S	788.53	y_3	brut
X_F	318.07	↑			
Y_F	788.44	GTO 1 9 R/S	–0.08	e_x	
		R/S	0.09	e_y	
		R/S	219.35	Σd	
$G_{0\to1}$	53.714	↑			
d	75.31	R/S	180.07	x_1	compensé
		R/S	801.35	y_1	compensé
$G_{1\to2}$	121.312	↑			
d	81.45	R/S	257,03	x_2	compensé
		R/S	774.56	y_2	compensé
$G_{2\to3}$	85.731	↑			
d	62.58	R/S	318.07	x_3	compensé
		R/S	788.44	y_3	compensé

INTERSECTION DE DROITES EN SÉRIE

Ce programme calcule les coordonnées X_i, Y_i du point d'intersection de deux droites.

Chaque droite peut être définie indifféremment par:

- deux points X_AY_A et X_BY_B ou par
- le gisement G^t et un point X_AY_A

Si une même droite est coupée par plusieurs autres, il suffit après avoir calculé le premier point d'intersection, d'introduire les éléments des droites sécantes variables.



Ce problème est résolu de la façon suivante; pour chaque droite, le programme calcule:

- la pente $a = \frac{\Delta y}{\Delta x}$ ou $a = \text{cotg}\, G^t$
- l'ordonnée à l'origine $b = Y_A - aX_A$

Les coordonnées du point d'intersection sont obtenues par:

$$X_i = \frac{b_2 - b_1}{a_1 - a_2} \qquad Y_i = a_1X_i + b_1$$

AFFICHAGE PAS	CODE	TOUCHES	AFFICHAGE PAS	CODE	TOUCHES
00			25	32	CHS
01	14 33	f REG	26	24 03	RCL 3
02	14 34	f STK	27	51	+
03	74	R/S	28	24 07	RCL 7
04	23 03	STO 3	29	15 61	g X≠0
05	22	R↓	30	13 36	GTO 36
06	23 02	STO 2	31	21	x⇄y
07	22	R↓	32	23 07	STO 7
08	21	x⇄y	33	24 04	RCL 4
09	15 71	g x=0	34	23 06	STO 6
10	13 18	GTO 18	35	13 02	GTO 02
11	24 02	RCL 2	36	41	–
12	41	–	37	24 06	RCL 6
13	21	x⇄y	38	24 04	RCL 4
14	24 03	RCL 3	39	41	–
15	41	–	40	71	÷
16	71	÷	41	74	R/S
17	13 21	GTO 21	42	24 06	RCL 6
18	21	x⇄y	43	61	×
19	15 34	g GRD	44	24 07	RCL 7
20	14 06	f TAN	45	51	+
21	15 22	g 1/x	46	74	R/S
22	23 04	STO 4	47	13 02	GTO 02
23	24 02	RCL 2	48		
24	61	X	49		

REGISTRES	
R_0	
R_1	
R_2	X_B
R_3	Y_B
R_4	a_2
R_5	
R_6	a_1
R_7	b_1

Exemple:

Calculer les coordonnées des points d'intersection de la droite ① connue par un point et le gisement, avec
la droite ② connue par deux points, et
la droite ③ connue par un point et le gisement.



Presser les touches [f] [PRGM] [R/S]

Pour la droite ① G^t 121.432 [↑]
X_A 500 [↑]
Y_A 800 [R/S] ⟶ Affichage de 0.00

Pour la droite ② X_B 850 [↑]
Y_B 912 [↑]
X_C 714 [↑]
Y_C 820 [R/S] ⟶ Affichage de $X_i = 621.55$
[R/S] ⟶ Affichage de $Y_i = 757.46$
[R/S] ⟶ Affichage de 0

Pour la droite ③ G^t 247.814 [↑]
X_D 1135 [↑]
Y_D 1218 [R/S] ⟶ Affichage de $X_i = 684.49$
[R/S] ⟶ Affichage de $Y_i = 735.43$
[R/S] ⟶ Affichage de 0

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Initialiser		f	PRGM	R/S		0.00
3	Pour la première droite	X_A	↑				
		Y_A	↑				
		X_B	↑				
		Y_B	R/S				0.00
	ou						
3'		G^t	↑				
		X_A	↑				
		Y_A	R/S				0.00
4	Pour chaque droite						
	sécante	X_A	↑				
		Y_A	↑				
		X_B	↑				
		Y_B	R/S				
	ou						
		G^t	↑				
		X_A	↑				
		Y_A	R/S				X_i
5	Lire X du point d'intersection		R/S				Y_i
6	Lire Y du point d'intersection						
7	Retourner en 4 ou 2 4=nouvelle droite						
	coupant la première 2=nouvelle intersection						
	indépendante des précédentes						

COTES PÉRIMÉTRIQUES, GISEMENTS, SURFACE D'UN POLYGONE

A partir de l'introduction des coordonnées successives X, Y des sommets d'un polygone, le programme calcule:

- les cotes périmétriques (affichés avec 2 décimales)
- les gisements des cotes (affichés avec 4 décimales)
- la surface du polygone (affichée avec 2 décimales).

(La surface du polygone est obtenue après réintroduction des coordonnées du premier sommet.)



Les formules utilisées sont les suivantes:

$$D = \sqrt{\Delta X^2 + \Delta Y^2}$$

$$G = \text{arc cos } \frac{\Delta Y}{D} \ (+400)$$

$$S = \frac{1}{2} \Sigma (X_{i+1} - X_i)(Y_{i+1} + Y_i)$$

Ce programme convient pour un nombre quelconque de sommets.

AFFICHAGE PAS	CODE	TOUCHES	AFFICHAGE PAS	CODE	TOUCHES
00			25	51	+
01	23 01	STO 1	26	14 11 04	f FIX 4
02	23 03	STO 3	27	74	R/S
03	22	R↓	28	21	x⇄y
04	23 00	STO 0	29	14 09	f R
05	23 02	STO 2	30	24 03	RCL 3
06	34	CLX	31	51	+
07	23 06	STO 6	32	23 03	STO 3
08	74	R/S	33	14 73	f last x
09	23 05	STO 5	34	51	+
10	22	R↓	35	61	×
11	23 04	STO 4	36	23 51 06	STO+6
12	24 02	RCL 2	37	24 00	RCL 0
13	41	−	38	24 04	RCL 4
14	24 05	RCL 5	39	23 02	STO 2
15	24 03	RCL 3	40	14 61	f x≠y
16	41	−	41	13 08	GTO 08
17	15 34	g GRD	42	24 01	RCL 1
18	15 09	g P	43	24 03	RCL 3
19	14 11 02	f FIX 2	44	14 61	f x≠y
20	74	R/S	45	13 08	GTO 08
21	21	x⇄y	46	24 06	RCL 6
22	15 51	g x⩾0	47	02	2
23	23 26	GTO 26	48	71	÷
24	24 07	RCL 7	49	13 19	GTO 19

REGISTRES	
R_0	X_1
R_1	Y_1
R_2	X_i
R_3	Y_i
R_4	X_{i+1}
R_5	Y_{i+1}
R_6	2 S
R_7	400

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire	400	STO	7			
3	Coordonnées du premier	X_1	↑				
	sommet	Y_1	f	PRGM	R/S		0.00
4	Coordonnées du point suivant	X_{i+1}	↑				
		Y_{i+1}	R/S				Distance
5	Lire la distance		R/S				Gisement
6	Lire le gisement						Gt ou S
7	Retourner en 4 ou		R/S				Surface
8	Lire la surface après						
	réintroduction du premier						
	sommet						

Exemples:



400 STO 7 (à faire dans tous les cas)

1	X 300	↑		
	Y 800	f PRGM R/S		
2	X 330	↑		
	Y 845	R/S	54.08	= D
		R/S	37.4334	= G 1→2
		R/S	330	= X_2
3	X 398	↑		
	Y 831	R/S	69.43	= D
		R/S	112.9263	= G 2→3
		R/S	398	= X_3
4	X 412	↑		
	Y 793	R/S	40.50	= D
		R/S	177.5279	= G 3→4
		R/S	412	= X_4
5	X 348	↑		
	Y 768	R/S	68.71	= D
		R/S	276.2924	= G 4→5
		R/S	348	= X_5
6	X 300	↑		
	Y 800	R/S	57.69	= D
		R/S	337.4334	= G 5→1
		R/S	5443,00	= Surface

CHAPITRE 8: TRIGONOMÉTRIE ET GÉOMÉTRIE ANALYTIQUE

TRANSFORMATION ET ROTATION D'AXES DE COORDONNÉES

Il est quelquefois nécessaire, en cartographie par exemple, d'avoir à effectuer une translation et/ou une rotation d'un système de coordonnées. L'origine est translatée de (0.0) à un nouveau point (x_0, y_0), et les axes x et y tournent ensuite d'un angle α, x′ et y′ étant les nouveaux axes.

Soit un point P de coordonnées (x, y) dans l'ancien système d'axe. Le problème consiste à trouver les nouvelles coordonnées (x′, y′) du point P dans un nouveau système d'axe.



Equations:

$$x' = (x - x_0)\cos\alpha + (y - y_0)\sin\alpha$$

$$y' = -(x - x_0)\sin\alpha + (y - y_0)\cos\alpha$$

Remarques:

1. Ce programme peut être utilisé pour résoudre soit un problème de translation, soit de rotation, ou bien à la fois de translation et de rotation. Dans le cas seulement d'une translation, introduire $\alpha = 0$. Dans le cas seulement d'une rotation, introduire $x_0 = y_0 = 0$.
2. Dans ce programme, il faut introduire α comme un nombre positif si la rotation s'effectue dans le sens inverse des aiguilles d'une montre ou comme un nombre négatif si la rotation s'effectue dans le sens des aiguilles d'une montre.

Remarque sur la programmation:

Ce programme est une application très intéressante de la conversion de coordonnées polaires/rectangulaires ([f] [→R]) associée aux possibilités des 4 registres de la pile opérationnelle. Les expressions $(x-x_0) \cos \alpha$, $(x-x_0) \sin \alpha$, $(y-y_0) \cos \alpha$ et $(y-y_0) \sin \alpha$ sont toutes obtenues par [f] [→R] et mises en mémoire dans la pile opérationnelle pour être utilisées ultérieurement. Un programme utilisant les touches [f] [sin] et [f] [cos] aurait nécessité 30 pas de mémoire programme (alors que celui-ci en occupe 19) et un registre mémoire supplémentaire.

AFFICHAGE		TOUCHES	X	Y	Z	T	COMMENTAIRES
PAS	CODE						
00			y	x			
01	23 03	STO 3	y	x			
02	22	R↓	x			y	
03	24 02	RCL 2	α	x			
04	21	x⇄y	x	α			
05	24 00	RCL 0	x_0	x	α		
06	41	–	Δx	α			$\Delta x = x - x_0$
07	14 09	f →R	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$			
08	24 03	RCL 3	y	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$		
09	24 01	RCL 1	y_0	y	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$	
10	41	–	Δy	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$	$\Delta x \sin \alpha$	$\Delta y = y - y_0$
11	24 02	RCL 2	α	Δy	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$	
12	21	x⇄y	Δy	α	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$	
13	14 09	f →R	$\Delta y \cos \alpha$	$\Delta y \sin \alpha$	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$	
14	22	R↓	$\Delta y \sin \alpha$	$\Delta x \cos \alpha$	$\Delta x \sin \alpha$	$\Delta y \cos \alpha$	
15	51	+	x'	$\Delta x \sin \alpha$	$\Delta y \cos \alpha$	$\Delta y \cos \alpha$	$x' = \Delta x \cos \alpha + \Delta y \sin \alpha$
16	74	R/S	x'	$\Delta x \sin \alpha$	$\Delta y \cos \alpha$	$\Delta y \cos \alpha$	
17	22	R↓	$\Delta x \sin \alpha$	$\Delta y \cos \alpha$	$\Delta y \cos \alpha$	x'	
18	41	–	y'	$\Delta y \cos \alpha$	x'	x'	$y' = -\Delta x \sin \alpha + \Delta y \cos \alpha$
19	13 00	GTO 00	y'	$\Delta y \cos \alpha$	x'	x'	
20							
21							
22							
23							
24							
25							
26							
27							
28							
29							
30							
31							
32							
33							
34							
35							
36							
37							
38							
39							
40							
41							
42							
43							
44							
45							
46							
47							
48							
49							

REGISTRES	
R_0	x_0
R_1	y_0
R_2	α
R_3	y
R_4	
R_5	
R_6	
R_7	

Exemple :

Un éclaireur doit trouver son chemin dans une forêt à la carte et à la boussole. Malheureusement, la carte n'est pas très pratique. Tout d'abord les cotes du quadrillage sont données en mètres, mais l'origine du plan se situe à plusieurs kilomètres, ce qui donne des nombres assez imposants. Ensuite la carte est cotée par rapport au Nord géographique alors que la boussole donne des indications par rapport au Nord magnétique (écart 17°).

Avant de partir, l'éclaireur décide de refaire une carte sommaire en prenant pour origine le point (54000, 118000) de l'ancienne carte et en faisant tourner les axes de 17° dans le sens des aiguilles d'une montre. Il doit trouver la position du pont et du sommet de la colline dans le nouveau système d'axes. Les anciennes coordonnées sont les suivantes :

Pont $\begin{cases} 55\,750 \\ 119\,300 \end{cases}$ Sommet $\begin{cases} 57\,450 \\ 120\,500 \end{cases}$



Solution :

54000 STO 0 118000 STO 1 17 CHS STO 2 f PRGM

55750 ↑ 119300 R/S ——→ 1293.45

R/S ——→ 1754.85

Les nouvelles coordonnées du pont sont (1293, 1755).

57450 ↑ 120500 R/S ——→ 2568.32

R/S ——→ 3399.44

Les nouvelles coordonnées du sommet de la colline sont (2568, 3399).

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire l'origine						
	du nouveau système de	x_0	STO	0			
	coordonnées	y_0	STO	1			
3	Mettre en mémoire l'angle de rotation	α	STO	2	f	PRGM	
4	Conversion de coordonnées						
	de l'ancien au nouveau	x	↑				
	système	y	R/S				x'
			R/S				y'
5	Effectuer 4 pour d'autres						
	points						
6	Pour un nouveau cas, aller en 2.						

RÉSOLUTION DU TRIANGLE (B, b, c)



Connaissant un angle (B), le côté opposé (b) et un des côtés adjacents (c), ce programme calcule au moyen des formules suivantes, les autres éléments du triangle:

1. $C = \text{arc sin}\left(\frac{c \sin B}{b}\right)$
2. $A = 2 \text{ arc sin } [1-(B+C)] = \pi - (B+C) = 180^\circ - (B+C) = 200 \text{ grades } -(B+C)$
3. $a = \frac{b \sin A}{\sin B}$

Si B est aigu (<90°) et b<c, il existe une deuxième solution qui se calcule au moyen des formules suivantes:

4. $C' = 2 \text{ arc sin } (1-C)$
5. $A' = 2 \text{ arc sin } [1-(B+C')]$
6. $a' = \frac{b \sin A'}{\sin B}$

Ce programme s'exécute dans n'importe quel mode angulaire (en mode DEGRÉS: degrés décimaux).

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	24 03	RCL 3
02	24 01	RCL 1
03	14 04	f SIN
04	61	x
05	24 02	RCL 2
06	71	÷
07	15 04	g SIN^{-1}
08	23 05	STO 5
09	74	R/S
10	24 01	RCL 1
11	51	+
12	01	1
13	15 04	g SIN^{-1}
14	02	2
15	61	x
16	23 04	STO 4
17	21	x⇄y
18	41	−
19	74	R/S
20	14 04	f SIN
21	24 02	RCL 2
22	61	x
23	24 01	RCL 1
24	14 04	f SIN

AFFICHAGE		TOUCHES
PAS	CODE	
25	71	÷
26	74	R/S
27	24 03	RCL 3
28	61	x
29	24 01	RCL 1
30	14 04	f SIN
31	61	x
32	02	2
33	71	÷
34	74	R/S
35	24 04	RCL 4
36	24 05	RCL 5
37	41	−
38	74	R/S
39	13 10	GTO 10
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1 B
R_2 b
R_3 c
R_4 2 arc sin 1
R_5 C
R_6
R_7

Exemple:

Soit: $B = 42.3°$
$b = 25.6$
$c = 32.8$

Calculer le triangle.

Solution:

B étant inférieur à 90° et b<c, deux solutions existent:

$C = 59.58°$
$A = 78.12°$
$a = 37.22$
Surface = 410.85

$C' = 120.42°$
$A' = 17.28°$
$a' = 11.30$
Surface = 124.68

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire B, b, c	B	STO	1			
		b	STO	2			
		c	STO	3			
3	Calcul du triangle		f	PRGM	R/S		C*
			R/S				A*
			R/S				a*
			R/S				Surface
4	Si B<90° et b<c, il existe						
	une deuxième solution		R/S				C'*
	*Après ce résultat, ne pas		R/S				A'*
	modifier le contenu de la pile.		R/S				a'*
			R/S				Surface'

RÉSOLUTION DU TRIANGLE (a, b, c)



Connaissant les trois côtés (a, b, c) d'un triangle, ce programme calcule, au moyen des formules suivantes, les trois angles:

$$C = \arc\cos\left(\frac{a^2 + b^2 - c^2}{2ab}\right)$$

$$B = \arc\sin\left(\frac{b \sin C}{c}\right) \qquad A = \arc\sin\left(\frac{a \sin C}{c}\right)$$

Modifier si nécessaire l'affectation des lettres pour que c désigne le plus grand côté. Le programme s'exécute dans n'importe quel mode angulaire (en mode DEGRÉS: degrés décimaux).

Ce programme calcule aussi la surface du triangle au moyen de la formule suivante:

$$\text{Surface} = \sqrt{s(s-a)(s-b)(s-c)}$$

où $s = \frac{1}{2}(a + b + c)$

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	24 01	RCL 1
02	24 02	RCL 2
03	15 09	g →P
04	15 02	g x^2
05	24 03	RCL 3
06	15 02	g x^2
07	41	−
08	24 01	RCL 1
09	24 02	RCL 2
10	61	x
11	02	2
12	61	x
13	71	÷
14	15 05	g COS^{-1}
15	74	R/S
16	14 04	f SIN
17	24 03	RCL 3
18	71	÷
19	23 00	STO 0
20	24 02	RCL 2
21	61	x
22	15 04	g SIN^{-1}
23	74	R/S
24	24 00	RCL 0

AFFICHAGE		TOUCHES
PAS	CODE	
25	24 01	RCL 1
26	61	x
27	15 05	g SIN^{-1}
28	74	R/S
29	24 01	RCL 1
30	24 02	RCL 2
31	51	+
32	24 03	RCL 3
33	51	+
34	02	2
35	71	÷
36	31	↑
37	23 00	STO 0
38	24 01	RCL 1
39	41	−
40	61	x
41	24 00	RCL 0
42	24 02	RCL 2
43	41	−
44	61	x
45	24 00	RCL 0
46	24 03	RCL 3
47	41	−
48	61	x
49	14 02	f $\sqrt{x}$

REGISTRES
R_0 Utilisé
R_1 a
R_2 b
R_3 c
R_4
R_5
R_6
R_7

Exemple:

Soit: a = 5.43, b = 10.46, c = 14.87

Solution:

C = 136.37°
B = 29.04°
A = 14.59°
Surface = 19.60

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire les côtés	a	STO	1			
	(c est le plus grand côté)	b	STO	2			
		c	STO	3			
3	Calcul du triangle		f	PRGM	R/S		C*
			R/S				B*
			R/S				A
			R/S				Surface
4	Calcul uniquement de la surface	a	STO	1			
		b	STO	2			
		c	STO	3			
			GTO	29	R/S		Surface
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

RÉSOLUTION DU TRIANGLE (a, A, C)



Connaissant deux angles (A, C) et un côté opposé (a), ce programme calcule au moyen des formules suivantes, les autres éléments du triangle:

$$B = 2\arcsin[1-(A+C)] = \pi-(A+C) = 180^\circ-(A+C) = 200\,\text{grades}-(A+C)$$

$$b = \frac{a \sin B}{\sin A}$$

$$c = \frac{a \sin C}{\sin A}$$

Le programme s'exécute dans n'importe quel mode angulaire (en mode DEGRÉS: degrés décimaux).

La surface est calculée au moyen de la formule suivante:

$$\text{Surface} = \frac{1}{2}\,ab \sin C$$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	01	1
02	15 04	g SIN^{-1}
03	02	2
04	61	x
05	24 02	RCL 2
06	24 03	RCL 3
07	51	+
08	41	−
09	74	R/S
10	14 04	f SIN
11	24 01	RCL 1
12	61	x
13	24 02	RCL 2
14	14 04	f SIN
15	71	÷
16	23 04	STO 4
17	74	R/S
18	24 01	RCL 1
19	14 73	f LASTx
20	71	÷
21	24 03	RCL 3
22	14 04	f SIN
23	61	x
24	74	R/S

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	24 01	RCL 1
26	24 04	RCL 4
27	61	x
28	24 03	RCL 3
29	14 04	f SIN
30	61	x
31	02	2
32	71	÷
33	13 00	GTO 00
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1 a
R_2 A
R_3 C
R_4 b
R_5
R_6
R_7

Exemple:

Soit a = 19.6, A = 40.25°, C = 61.06°

Solution:

B = 78.69°

b = 29.75

c = 26.55

Surface = 255.11

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire a, A et C	a	STO	1			
		A	STO	2			
		C	STO	3			
3	Calcul du triangle		f	PRGM	R/S		B*
			R/S				b*
			R/S				c
			R/S				Surface
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

RÉSOLUTION DU TRIANGLE (a, b, C)



Connaissant les deux côtés (a, b) et l'angle de ces deux côtés (C), ce programme calcule au moyen des formules suivantes, les autres éléments du triangle:

$c = \sqrt{a^2 + b^2 - 2ab\cos C}$ $\qquad A = \arc\sin\left(\frac{a \sin C}{c}\right)$

$B = 2\arc\sin[1-(A+C)] = \pi-(A+C) = 180^\circ-(A+C) = 200\text{ grades} -(A+C)$

La surface est calculée par la formule suivante:

$\text{Surface} = \frac{1}{2}\, ab \sin C$

Modifier si nécessaire l'affectation des lettres pour que a soit inférieur à b.

Ce programme s'exécute dans n'importe quel mode angulaire (en mode DEGRÉS: degrés décimaux).

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	24 01	RCL 1
02	24 02	RCL 2
03	15 09	g →P
04	15 02	g x^2
05	24 01	RCL 1
06	24 02	RCL 2
07	61	x
08	02	2
09	61	x
10	24 03	RCL 3
11	14 05	f COS
12	61	x
13	41	–
14	14 02	f $\sqrt{x}$
15	74	R/S
16	24 01	RCL 1
17	24 03	RCL 3
18	14 04	f SIN
19	61	x
20	21	x⇄y
21	71	÷
22	15 04	g SIN^{-1}
23	74	R/S
24	01	1
25	15 04	g SIN^{-1}
26	02	2
27	61	x
28	21	x⇄y
29	24 03	RCL 3
30	51	+
31	41	–
32	74	R/S
33	24 03	RCL 3
34	14 04	f SIN
35	24 01	RCL 1
36	61	x
37	24 02	RCL 2
38	61	x
39	02	2
40	71	÷
41	13 00	GTO 00
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1 a
R_2 b
R_3 C
R_4
R_5
R_6
R_7

Exemple:

Soit: a = 146
b = 227
C = 31.49°

Solution:

c = 127.76
A = 36.65°
B = 111.86°
Surface = 8655.86

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire a, b et C						
	(a<b)	a	STO	1			
		b	STO	2			
		C	STO	3			
3	Calcul du triangle		f	PRGM	R/S		c*
			R/S				A*
			R/S				B
			R/S				Surface
4	Calcul uniquement de la surface	a	STO	1			
		b	STO	2			
		C	STO	3			
			GTO	33	R/S		Surface
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

RÉSOLUTION DU TRIANGLE (a, B, C)



Connaissant deux angles (B, C) et leur côté commun (a), ce programme calcule au moyen des formules suivantes les autres éléments du triangle:

$A = 2 \text{ arc sin } [1-(B+C)] = \pi-(B+C) = 180°-(B+C) = 200 \text{ grades} -(B+C)$

$b = \dfrac{a \sin B}{\sin A} \quad c = \dfrac{a \sin C}{\sin A}$

La surface est calculée par la formule:

$$\text{Surface} = \frac{a^2 \sin B \sin C}{2 \sin (B+C)}$$

Ce programme s'exécute dans n'importe quel mode angulaire (mode DEGRÉS: degrés décimaux).

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	01	1
02	15 04	g SIN^{-1}
03	02	2
04	61	x
05	24 02	RCL 2
06	24 03	RCL 3
07	51	+
08	41	−
09	23 04	STO 4
10	74	R/S
11	24 01	RCL 1
12	24 04	RCL 4
13	14 04	f SIN
14	71	÷
15	23 04	STO 4
16	24 02	RCL 2
17	14 04	f SIN
18	61	x
19	74	R/S
20	24 04	RCL 4
21	24 03	RCL 3
22	14 04	f SIN
23	61	x
24	74	R/S

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25	24 01	RCL 1
26	15 02	g x^2
27	02	2
28	71	÷
29	24 02	RCL 2
30	14 04	f SIN
31	61	x
32	24 03	RCL 3
33	14 04	f SIN
34	61	x
35	24 02	RCL 2
36	24 03	RCL 3
37	51	+
38	14 04	f SIN
39	71	÷
40	13 00	GTO 00
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1 a
R_2 B
R_3 C
R_4 A, (a/sin A)
R_5
R_6
R_7

Exemple:

Soit: a = 20.96
B = 64°32′
C = 35°06′

Solution:

Convertir d'abord les angles B et C en degrés décimaux.
A = 80.37°
b = 19.19
c = 12.22
Surface = 115.66

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	Mettre en mémoire a, B, C	a	STO	1			
		B	STO	2			
		C	STO	3			
3	Calcul du triangle		f	PRGM	R/S		A*
			R/S				b*
			R/S				c
			R/S				Surface
4	Calcul uniquement de la surface	a	STO	1			
		B	STO	2			
		C	STO	3			
			GTO	25	R/S		Surface
	*Après ce résultat, ne pas						
	modifier le contenu de la pile.						

FONCTIONS HYPERBOLIQUES

Ce programme calcule les six fonctions hyperboliques au moyen des formules suivantes:

1. $\text{sh } x = \dfrac{e^x - e^{-x}}{2}$
2. $\text{ch } x = \dfrac{e^x + e^{-x}}{2}$
3. $\text{th } x = \dfrac{e^x - e^{-x}}{e^x + e^{-x}}$
4. $\text{csch } x = \dfrac{1}{\text{sh } x} \quad (x \neq 0)$
5. $\text{sech } x = \dfrac{1}{\text{ch } x}$
6. $\text{coth } x = \dfrac{1}{\text{th } x} \quad (x \neq 0)$

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
00		
01	15 07	g e^x
02	31	↑
03	15 22	g 1/x
04	41	−
05	02	2
06	71	÷
07	13 00	GTO 00
08	15 07	g e^x
09	31	↑
10	15 22	g 1/x
11	51	+
12	13 05	GTO 05
13	15 07	g e^x
14	31	↑
15	15 22	g 1/x
16	41	−
17	31	↑
18	31	↑
19	14 73	f LASTx
20	02	2
21	61	×
22	51	+
23	71	÷
24	13 00	GTO 00

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES
25		
26		
27		
28		
29		
30		
31		
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1
R_2
R_3
R_4
R_5
R_6
R_7

Exemples:

1. sh 2.5 = 6.05
2. ch 3.2 = 12.29
3. th 1.9 = 0.96
4. csch 4.6 = 0.02
5. sech (–0.25) = 0.97
6. coth (–2.01) = –1.04

N°	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	sh x	x	f	PRGM	R/S		sh x
	ou						
	ch x	x	GTO	08	R/S		ch x
	ou						
	th x	x	GTO	13	R/S		th x
	ou						
	csch x	x	f	PRGM	R/S		
			g	$^1/_x$			csch x
	ou						
	sech x	x	GTO	08	R/S		
			g	$^1/_x$			sech x
	ou						
	coth x	x	GTO	13	R/S		
			g	$^1/_x$			coth x

FONCTIONS HYPERBOLIQUES INVERSES

Ce programme calcule les fonctions hyperboliques inverses au moyen des formules suivantes:

1. $\arg \mathrm{sh}\ x = \ln [x + (x^2 + 1)^{1/2}]$
2. $\arg \mathrm{ch}\ x = \ln [x + (x^2 - 1)^{1/2}]$ $\qquad x \geqslant 1$
3. $\arg \mathrm{th}\ x = \frac{1}{2} \ln \left[\frac{1+x}{1-x}\right]$ $\qquad x^2 < 1$
4. $\mathrm{arc\ csch}\ x = \arg \mathrm{sh} \left[\frac{1}{x}\right]$ $\qquad x \neq 0$
5. $\mathrm{arc\ sech}\ x = \arg \mathrm{ch} \left[\frac{1}{x}\right]$ $\qquad 0 < x \leqslant 1$
6. $\arg \mathrm{coth}\ x = \arg \mathrm{th} \left[\frac{1}{x}\right]$ $\qquad x^2 > 1$

AFFICHAGE		TOUCHES
PAS	CODE	
00		
01	31	↑
02	31	↑
03	61	x
04	01	1
05	51	+
06	14 02	f $\sqrt{x}$
07	51	+
08	14 07	f LN
09	13 00	GTO 00
10	31	↑
11	31	↑
12	61	x
13	01	1
14	41	–
15	14 02	f $\sqrt{x}$
16	51	+
17	14 07	f LN
18	13 00	GTO 00
19	31	↑
20	31	↑
21	01	1
22	51	+
23	21	x⇄y
24	32	CHS

AFFICHAGE		TOUCHES
PAS	CODE	
25	01	1
26	51	+
27	71	÷
28	14 07	f LN
29	02	2
30	71	÷
31	13 00	GTO 00
32		
33		
34		
35		
36		
37		
38		
39		
40		
41		
42		
43		
44		
45		
46		
47		
48		
49		

REGISTRES
R_0
R_1
R_2
R_3
R_4
R_5
R_6
R_7

Exemples:

1. arg sh (2.4) = 1.61
2. arg ch (90) = 5.19
3. arg th (−0.65) = −0.78
4. arc csch (2) = 0.48
5. arc sech (0.4) = 1.57
6. arg coth (3.4) = 0.30

Nº	INSTRUCTIONS	DONNÉES	TOUCHES				RÉSULTATS
1	Introduire le programme						
2	arg sh x	x	f	PRGM	R/S		arg sh x
	ou						
	arg ch x	x	GTO	10	R/S		arg ch x
	ou						
	arg th x	x	GTO	19	R/S		arg th x
	ou						
	arc csch x	x	f	PRGM	R/S		
			g	1/x			arc csch x
	ou						
	arc sech x	x	GTO	10	R/S		
			g	1/x			arc sech x
	ou						
	arg coth x	x	GTO	19	R/S		
			g	1/x			arg coth x

HEWLETT PACKARD

FEUILLE DE PROGRAMMATION POUR LE HP-25

Titre ______________________ Page ______

Appuyez sur BST en mode RUN, passez en mode PRGM, puis introduisez votre programme.

AFFICHAGE PAS	AFFICHAGE CODE	TOUCHES	X	Y	Z	T	COMMENTAIRES
00							
01							
02							
03							
04							
05							
06							
45							
46							
47							
48							
49							

REGISTRES
R_0
R_1

INDEX